



A MARSHALL CAVENDISH

6

COMPUTER COURSE IN WEEKLY PARTS

INPUT

LEARN PROGRAMMING - FOR FUN AND THE FUTURE

0 1 1A 4E 68 7F
FF E6 D4 98 81

UK £1.00 Republic of Ireland £1.25 Malta 85c Australia \$2.25 New Zealand \$2.95

INPUT

Vol 1

No 6

GAMES PROGRAMMING 6

CREATING THE BIG BANG

161

Routines you can add to a variety of games to create convincing screen 'explosions'

BASIC PROGRAMMING 11

SPRITES ON THE COMMODORE 64

168

A special, and important, graphics feature of these machines

BASIC PROGRAMMING 12

GET YOUR PROGRAMS IN SHAPE

173

The first of two articles explaining how to write better structured, smoother running programs

MACHINE CODE 7

GETTING DOWN BELOW ZERO

179

How to deal with negative numbers in hex and binary arithmetic – and when you'll need them

BASIC PROGRAMMING 13

REFINING YOUR SCREEN GRAPHICS

184

Getting to grips with more sophisticated uses of the BASIC graphics commands

INDEX

The last part of INPUT, Part 52, will contain a complete, cross-referenced index. For easy access to your growing collection, a cumulative index to the contents of each issue is contained on the inside back cover.

PICTURE CREDITS

Front cover, Dave King. Page 161, British Aerospace. Pages 162, 164, 166, Chris Lyons. Page 168, Tudor Art Studio/Bernard Robinson. Page 170, Bernard Fallon. Page 172, Howard Kingsnorth. Pages 174, 176, Nick Farmer. Page 178, Nick Farmer, Howard Kingsnorth. Page 180, Dave King. Page 182, David Lloyd, Dick Ward. Page 184, Hussein Hussein, Martin Cleaver. Page 186, Hussein Hussein, Peter Dazely. Page 188, David Lloyd. Page 190, Hussein Hussein, The Photographers Library.

© Marshall Cavendish Limited 1984/5/6

All worldwide rights reserved.

The contents of this publication including software, codes, listings, graphics, illustrations and text are the exclusive property and copyright of Marshall Cavendish Limited and may not be copied, reproduced, transmitted, hired, lent, distributed, stored or modified in any form whatsoever without the prior approval of the Copyright holder.

Published by Marshall Cavendish Partworks Ltd, 58 Old Compton Street, London W1V 5PA, England. Printed by Artisan Press, Leicester and Howard Hunt Litho, London.

HOW TO ORDER YOUR BINDERS

UK and Republic of Ireland: Send £4.95 (inc p & p) (IR£5.45) for each binder to the address below:

Marshall Cavendish Services Ltd,
Department 980, Newtown Road,
Hove, Sussex BN3 7DN

Australia: See inserts for details, or write to INPUT, Gordon and Gotch Ltd, PO Box 213, Alexandria, NSW 2015

New Zealand: See inserts for details, or write to INPUT, Gordon and Gotch (NZ) Ltd, PO Box 1595, Wellington

Malta: Binders are available from local newsgents.



BACK NUMBERS

Copies of any part of INPUT can be obtained from the following addresses at the regular cover price, with no extra charge for postage and packing:

UK and Republic of Ireland:

INPUT, Dept AN, Marshall Cavendish Services,
Newtown Road, Hove BN3 7DN

Australia, New Zealand and Malta:

Back numbers are available through your local newsgent

COPIES BY POST

Our Subscription Department can supply your copies direct to you regularly at £1.00 each. For example the cost of 26 issues is £26.00; for any other quantity simply multiply the number of issues required by £1.00. These rates apply anywhere in the world. Send your order, with payment to:

Subscription Department, Marshall Cavendish Services Ltd,
Newtown Road, Hove, Sussex BN3 7DN

Please state the title of the publication and the part from which you wish to start.

HOW TO PAY: Readers in UK and Republic of Ireland: All cheques or postal orders for binders, back numbers and copies by post should be made payable to:

Marshall Cavendish Partworks Ltd.


QUERIES: When writing in, please give the make and model of your computer, as well as the Part No., page and line where the program is rejected or where it does not work. We can only answer specific queries – and please do not telephone. Send your queries to INPUT Queries, Marshall Cavendish Partworks Ltd, 58 Old Compton Street, London W1V 5PA.

INPUT IS SPECIALLY DESIGNED FOR:

The SINCLAIR ZX SPECTRUM (16K, 48K, 128 and +),
COMMODORE 64 and 128, ACORN ELECTRON, BBC B
and B+, and the DRAGON 32 and 64.

In addition, many of the programs and explanations are also suitable for the SINCLAIR ZX81, COMMODORE VIC 20, and TANDY COLOUR COMPUTER in 32K with extended BASIC. Programs and text which are specifically for particular machines are indicated by the following symbols:

 **SPECTRUM 16K, 48K, 128, and +**  **COMMODORE 64 and 128**

 **ACORN ELECTRON, BBC B and B+**  **DRAGON 32 and 64**

 **ZX81**  **VIC 20**  **TANDY TRS80 COLOUR COMPUTER**

CREATING THE BIG BANG

Explosions are a standard part of the games programmers' repertoire—in anything from aerial combat to interstellar wars. Here are some ways to program suitable visual effects that you can use for a whole variety of games.

It's quite easy to make your games a lot more spectacular by adding special effects graphics routines. And they need not be very complex programs to make a big difference to the look of your games.

As you'll see, there are many different ways to produce visual effects—all you have to remember is to use the right sort of effect in the right sort of game.

The flame effects detailed here are suitable for all sorts of games involving buildings, cars, ships, tanks and so on, but wouldn't really look right in a space game.

Also, as the flames and explosions have been drawn using UDGs, the maximum target size is limited. And this means that you can't just add them to any game, without altering the routine.

On the other hand, some of the machines have screen flash routines which are much better for space games, and have the added advantage that they need no adaptation for use in any game.

- CREATING SCREEN FLASHES
- A SIMPLE AEROPLANE BOMBING
RUN PROGRAM
- HOW TO ADD FLAMES AND
EXPLOSIONS

The key to the Spectrum 'explosion' is a POKE which makes the fire and debris of a bomb blast first flare up, then gradually vanish from the screen as the building collapses. This is explained more fully below.

But first you need an aircraft—and a bomb. So enter this program:

```
10 FOR n=USR "p" to USR "q" + 7
20 READ a
30 POKE n,a
40 NEXT n
50 DATA 32,16,136,154,155,8,16,32
60 DATA 0,16,16,120,28,28,0,0
```

This uses conventional user defined graphics, as described on pages 38 to 45, to create images for both the aircraft and bomb. RUN the program and then check you have



the screen is completed each time.

If you don't like the stripey effect of the last program then add a short delay which will allow the complete screen to change colour:

```
8025 FOR delay = 1 TO 200 : NEXT
```

Change Line 8000 as well, or the flashing will last too long:

```
8000 FOR T = 1 TO 3
```

If you want to use the program as a subroutine in a game, then you'll have to make two alterations. First, delete Line 7090—as the mode will already be defined in the main program. Then add a RETURN line, such as 8060 RETURN, to the end of the program.

If you SAVED the alien game from Games Programming 5 then you can use that to try out the effect. Otherwise, use the simple

animation program below. It shows an alien being hit by a missile:

```
10 MODE1
15 VDU23;8202;0;0;0;
17 CLS
20 VDU23,224,60,126,219,219,126,60,
  92,153
30 VDU23,225,16,16,16,56,56,56,40,108
40 PRINTTAB(19,4)CHR$(224)
50 FOR Y = 31 TO 3 STEP -1
60 PRINTTAB(19,Y)CHR$(225)
70 A = INKEY(10)
80 PRINTTAB(19,Y)“□”;
90 NEXT
100 GOSUB 8000
110 GOTO 17
8060 RETURN
```

Don't forget that there must always be a line in the main program to call the subroutine. In the alien program above, this is:

```
100 GOSUB 8000
```

So add a line like this (with the appropriate line numbers) whenever you want to use the subroutine in your own programs.

GROUND EXPLOSIONS

So far the program has manipulated the screen display as a whole. An alternative is to try writing some suitable animated graphics.

Here's an idea which animates some flames. They are meant to be superimposed over a target which has been successfully destroyed. You could use the flames in a game involving tanks, ships, buildings, or any ground-based target.

The program to animate the flames is written as a PROCEDURE so it can be added to any suitable program.

```
10 MODE1
20 VDU 5
30 FOR T = 224 TO 229:VDU23,T
40 FOR P = 0 TO 7:READ A:VDU A
50 NEXT P
60 NEXT T
70 GCOL0,2:MOVE0,100:MOVE 0,200:PLOT
  85,1280,100:PLOT 85,1280,200
80 Q = 1:S = 1:X = 600:Y = 232:FY = Y:
  BY = 763
90 PROCFACTORY
100 FOR T = 0 TO 1200 STEP 16
110 MOVE T,800:GCOL0,3:VDU229
120 IF T > 128 AND BY > 250 THEN
  PROCBOMB
130 IF BY <= 250 AND Q < 16 THEN
  PROCEXPLOSION
140 MOVE T,800:GCOL0,0:VDU226
150 NEXT T
160 GOTO 80
```

```
170 DEF PROCBOMB
180 MOVE T - 16,BY:GCOL0,0:VDU226
190 BY = BY - 16
200 MOVE T,BY:GCOL0,3:VDU228
210 ENDPROC
220 DEF PROCFACTORY
230 GCOL0,3:MOVE 600,200:MOVE692,200:
  PLOT85,600,231:PLOT85,692,231:
  GCOL0,2:MOVE600,263
240 VDU227,227,227
250 ENDPROC
8000 DEF PROCEXPLOSION
8010 GCOL0,0:MOVE X,Y:VDU226,226,226
8020 MOVE X,Y + 32:VDU226,226,226
8030 GCOL0,2:MOVE X,FY
8040 VDU224 + S,224 + S,224 + S
8050 S = 1 - S
8060 FY = FY - 2
8070 Q = Q + 1
8080 ENDPROC
9000 DATA 2,128,25,126,126,255,255,255,
  4,33,144,66,231,255,255,255
9010 DATA 255,255,255,255,255,255,255,
  255
9020 DATA 0,0,0,0,3,15,63,255,0,16,16,
  120,28,28,0,0
9030 DATA 32,16,136,254,255,8,16,32
```

This program shows a plane dropping a bomb on a factory. If the factory is destroyed, then two sets of flames are PRINTed, one after the other to give the impression of a flickering fire. The flames also gradually die down as the program proceeds.

Here's how the program works, line-by-line to create the effect.

The DATA for the two sets of flames is in Line 9000 and the other DATA lines are for a blank square, the plane and the bomb. Lines 30 to 60 READ the DATA and set up the UDGs, and the flames are PRINTed, at the correct position, by Lines 8030 and 8040. Lines 8010 and 8020 simply blank out the factory before the flames are displayed. Which set of flames is PRINTed by Line 8040 depends on the value of S. S starts off equal to 1, set in Line 80, so the first time the PROCEDURE is called it PRINTs character 244 + 1, that is 245. Line 8050 changes the value of S to 0 so the next time round, character 244 + 0, i.e. 244, is PRINTed. The third time round, S is changed back to 1 again and so on. As the characters are swapped back and forth the flames flicker and appear to be animated. Line 8060 alters the Y co-ordinate of the flames making them move down the screen each time giving the impression that they are dying down.

Finally, Line 8070 uses Q to keep count of how many times the flames are PRINTed and when Q = 16 the condition in Line 130 becomes false so the procedure is not called



again until the next bomb drops.

The rest of the program controls the movement of the plane and the bomb and draws the screen display. Line 70 PLOTS the ground at the bottom of the screen, Line 90 DRAWs the factory, and Lines 100, 110 and 140 move the plane across the top of the screen. PROCBOMB is called in Line 120 but only if the plane has reached its target at position 250. When the bomb does reach its target then PROCXPLOSION is called instead by Line 130.



Dramatic visual effects that will greatly enhance space games can be achieved on the Dragon or Tandy with very short programs. Try typing in this one and RUNNING it:

```
7980 PMODE 3,1
7990 PCLS
8000 FOR F=1 TO 1000
8010 SCREEN 1,0
8020 SCREEN 1,1
8030 NEXT F
8040 CLS
```

You'll see bands of colour creeping down the screen, caused by the computer switching very quickly between colour sets. It's an effect you could use for the end of a particular phase of a game—penetration of a shield, perhaps?

The mode is set up by Line 7980 and then the graphics screen is cleared by Line 7990. Line 8010 switches on one colour set, whilst Line 8020 switches back to the other. The FOR . . . NEXT loop in Lines 8000 and 8030 switches the two colour sets 1000 times.

This switching changes the colour sets very rapidly—so quickly, in fact, that the TV never has enough time to complete changing the colour of the screen before it has to swap it back again. Only a small band of the screen is thus completed in each colour.

If you want to use the program as a subroutine—either with a game (it'll only work with games written in one of the PMODEs, such as that on page 144) or with the alien program given below—you'll have to

make a few alterations. First, delete Lines 7980 and 7990, then add a RETURN line such as 8500 RETURN.

```
10 PMODE 1,1
20 DIM A(3),B(3),M(3)
200 FOR K=1536 TO 2016 STEP 32
210 READ A,B:POKE K,A:POKE K+1,B
220 NEXT
230 GET(0,0) — (15,15),A,G
240 GET(0,16) — (15,31),M,G
250 PCLS
260 MX=120:MY=191:PX=120:PY=20
270 PUT(PX,PY) — (PX+15,PY+15),
    A,PSET
280 SCREEN 1,0
290 PUT (MX,MY) — (MX+15,MY+15),
    B,PSET
300 MY=MY-4
310 IF MY < 36 THEN GOSUB 8000:
    GOTO 260
320 PUT (MX,MY) — (MX+15,MY+15),
    M,PSET
330 GOTO 290
9000 DATA 252,63,3,192,15,240,61,124,
    58,172,245,95,213,87,213,87
9010 DATA 0,128,0,128,0,128,2,160,10,
    168,0,192,3,240,15,60
```

The program shows an alien being hit by a missile. When the alien is hit, Line 310 calls up the screen stripe routine.

When used as a subroutine, the screen stripe program will make stripes in any PMODE you like, so it can be added to any game without altering Lines 8000 to 8040.

Here's an adaptation of the above stripes program which gives a more elaborate effect, although the screen graphics won't remain intact as they did in the first case. You can type it in and RUN it as it is, or add it to another program—such as the alien animation—as a subroutine. If you use it as a subroutine, you'll need to delete Line 7990 and add a RETURN line—as you did before.

```
7990 PMODE 3,1
8000 FOR F=1 TO 3
8010 FOR K=0 TO 1
```

```
8020 SCREEN 1,K
8030 FOR J=1 TO 4
8040 PCLS J
8050 NEXT J
8060 NEXT K
8070 NEXT F
8080 CLS
```

As a subroutine it will work in both the two-colour modes (PMODEs 0, 2 and 4), and the four-colour modes (PMODEs 1 and 3). In addition to changing the colour sets, the routine also clears the screen and changes the screen colour.

Lines 8030 to 8050 are a FOR . . . NEXT loop which turns the screen to each of the available colours in the colour set in turn. The program is designed to look for four colours in the set, but will work in a two-colour mode despite this. No errors will be reported and only two colours will appear on the screen.

One last variation on the program looks like this:

```
7990 PMODE 3,1
8000 FOR F=1 TO 5
8010 SCREEN 1,0
8020 FOR K=1 TO 200: NEXT K
8030 SCREEN 1,1
8040 FOR K=1 TO 200: NEXT K
8050 NEXT F
8060 CLS
```

Either type it in as it is, or use it as a subroutine by deleting Line 7990 and adding a RETURN line at the end.

You will see the screen flashing in colour. This happens because Lines 8020 and 8040 insert short delays allowing the complete screen to change colour before the colour sets are swapped. You can experiment by altering the length of the delay and the number of times that the colours are swapped.

FLAMES

So far you've seen some ideas for manipulating the screen display as a whole. But even better visual effects can be achieved by writing some suitable animated graphics.

Here's an idea which animates some flames

that you could superimpose over a target which has been successfully destroyed. Alternatively, you could use them in a game involving any ground-based target.

The program has been written in PMODE 1, and can't be used as it stands with games written in other PMODEs. Type it in and RUN it to see what it looks like:

```

10 PMODE 1,1
20 DIM B(3),E1(3),E2(3)
30 FOR K=1536 TO 2016 STEP 32
40 READ A,B:POKE K,A:POKE K+1,B
50 NEXT
60 GET (0,0) - (15,15),E1,G
70 GET (0,16) - (15,31),E2,G
80 SCREEN 1,0
250 PCLS:HX=124:HY=146
8000 FOR N=0 TO 15
8010 PUT (HX,HY+N) - (HX+15,HY+15),
    E1,PSET
8020 FOR K=1 TO 100:NEXT
8030 PUT (HX,HY+N) - (HX+15,HY+15),
    E2,PSET
8040 FOR K=1 TO 100:NEXT
8050 PUT (HX,HY+N) - (HX+15,HY+15),
    B,PSET
8060 NEXT
9000 DATA 0,12,192,0,3,195,63,252,
    63,252,255,255,255,255,255,255
9010 DATA 0,48,12,3,195,0,48,12,252,
    63,255,255,255,255,255,255

```

Two sets of flames are PUT on the screen, one after the other, giving the impression of fire. The flames also gradually die down.

The program works like this:

The DATA for the flames is in Lines 9000 and 9010. The flames are POKed on to the screen by Lines 30 to 50. Because the program is written in a four-colour mode the DATA is a little different from the DATA you saw in Machine Code 2 (page 40). Setting up colour UDGs will be dealt with in detail later.

Lines 60 to 70 GET the two flame shapes into arrays E1 and E2—which were DIMensioned in Line 20.

Lines 8000 to 8060 animate the flames and make them die down. Each time the program goes through the FOR ... NEXT loop the two flame arrays are PUT on the screen, and blanked out by PUTting the blank array, B, on top of them. The top of the flames is lowered each time by the +N in the PUT lines, so the flames appear to be dying down. If you don't see how this works try plotting the coordinates in the PUT lines as the program goes through the FOR ... NEXT loop.

Although the program will work on its own, it'll look a lot better if you try it out as a subroutine with the bombing run program given below—you'll see a building being destroyed and then burned down. If you use the two programs together make sure that Line 20 is altered so that it appears in the bombing run—use the machine's editor, or retype the line. Also, delete Line 80—just type 80 [ENTER].

```

20 DIM A(3),B(3),H(3),E1(3),E2(3)
200 FOR K=1536 TO 2016 STEP 32
210 READ A,B:POKE K,A:POKE K+1,B
220 NEXT
230 GET(0,0) - (15,15),A,G
240 GET(0,16) - (15,31),H,G
250 PCLS:LINE(0,163) - (255,191),
    PSET,BF
260 HX=124:HY=146:PX=0:PY=40:B=0
270 PUT(HX,HY) - (HX+15,HY+15),H,PSET
280 SCREEN 1,0
290 PUT(PX,PY) - (PX+15,PY+15),B,PSET
300 PX=PX+4
310 PUT(PX,PY) - (PX+15,PY+15),A,
    PSET
320 IF PX=20 THEN B=1:BX=PX+8:
    BY=PY+8
330 IF B=1 THEN PRESET(BX,BY):
    PRESET(BX+2,BY):BX=BX+2:BY=
    BY+2:PSET(BX,BY,4):PSET(BX+2,BY,4)
340 IF BY=148 THEN GOSUB 8000:BY=0:
    GOTO 250
350 GOTO 290
8070 RETURN

```

```

9020 DATA 0,0,2,0,130,128,162,160,170,
    170,162,160,130,128,2,0
9030 DATA 0,3,12,51,60,243,255,255,
    255,255,85,85,86,149,86,149

```

If you want to use the subroutine in other PMODEs it will have to be changed slightly. Here are two versions which will allow the flames to be used with games written in other modes.

First, here is a program suitable for use in PMODEs 3 and 4—change the mode number as appropriate:

```

10 PMODE 3,1
20 DIM B(6),E1(6),E2(6)
30 FOR K=1536 TO 2496 STEP 64
40 READ A,B:POKE K,A:POKE K+1,B
45 POKE K+32,A:POKE K+33,B
50 NEXT
60 GET(0,0) - (15,15),E1,G
70 GET(0,16) - (15,31),E2,G
80 SCREEN 1,0
250 PCLS:HX=124:HY=146
8000 FOR N=0 TO 15
8010 PUT(HX,HY+N) - (HX+15,HY+15),
    E1,PSET
8020 FOR K=1 TO 100:NEXT
8030 PUT(HX,HY+N) - (HX+15,HY+15),
    E2,PSET
8040 FOR K=1 TO 100:NEXT
8050 PUT(HX,HY+N) - (HX+15,HY+15),
    B,PSET
8060 NEXT
9000 DATA 0,12,192,0,3,195,63,252,63,
    252,255,255,255,255,255
9010 DATA 0,48,12,3,195,0,48,12,252,63,
    255,255,255,255,255,255


```

And secondly, here is a program suitable for use in PMODE 2.

```

10 PMODE 2,1
20 DIM B(6),E1(6),E2(6)
30 FOR K=1536 TO 2496 STEP 32
40 READ A:POKE K,A:POKE K+16,A
50 NEXT
60 GET(0,0) - (15,15),E1,G
70 GET(0,16) - (15,31),E2,G
80 SCREEN 1,0
250 PCLS:HX=124:HY=146
8000 FOR N=0 TO 15
8010 PUT(HX,HY+N) - (HX+15,HY+15),
    E1,PSET
8020 FOR K=1 TO 100:NEXT
8030 PUT(HX,HY+N) - (HX+15,HY+15),
    E2,PSET
8040 FOR K=1 TO 100:NEXT
8050 PUT(HX,HY+N) - (HX+15,HY+15),
    B,PSET
8060 NEXT
9000 DATA 2,128,25,126,126,255,255,255
9010 DATA 4,33,144,66,231,255,255,255

```



SPRITES ON THE COMMODORE 64

The sprite is a standard graphics facility on the Commodore 64, producing a very mobile high resolution image. Simple to master, it's the basis of many games.

A sprite—also called a movable object block (MOB)—is a special kind of highly mobile high resolution UDG some uses of which have already been seen. Apart from its mobility, it has other unusual characteristics—for example, the ability to expand and contract in width and height upon command. It is not therefore surprising to find that sprites are very commonly used in games programming—but, in fact, sprites can be used in any program where high resolution animated graphics are required. A business program, for example, could make use of one or more sprites to form an 'icon' (symbol depicting choice of program operation), or as part of an animated logo.

There are two types of sprite: standard high resolution form or multicolour. The difference between the two is that the former adopt the chosen foreground colour. Multicolour sprites offer a choice of up to four colours simultaneously, but only by sacrificing some of the horizontal resolution.

To start with, let's look at standard sprites. Multicolour sprites are covered separately.

DEFINING A SPRITE

Sprites are defined very much like standard UDGs (see page 38). But a sprite is larger and so needs more DATA to define it. In fact, it is three times the width of a standard UDG, and not quite three times its depth, occupying an area 24 pixels by 21. Not all of the 504 available pixel positions have to be used (that is, turned on)—and it may not be possible to do so in certain instances.

Rather than defining each of the 24 pixels on each row separately, information about the sprite shape is contained in three groups of eight (three bytes) on each of the 21 rows. This gives a key to the way DATA statement values used for normal UDGs can be adapted for use in this case, since these are also calculated in groups of eight.

The arrangement of bytes in the sprite takes the following form:

Row 1:	BYTE 1	BYTE 2	BYTE 3
Row 2:	BYTE 4	BYTE 5	BYTE 6
Row 3:	BYTE 7	BYTE 8	BYTE 9

... and continues up to:



Row 20:	BYTE 58	BYTE 59	BYTE 60
Row 21:	BYTE 61	BYTE 62	BYTE 63

So you have side by side groupings of three bytes for each of the 21 sprite rows.

Each of these bytes handles information in the same way as a single line of the UDG on page 38. If you think in terms of decimal notation, for instance, the possible values for the pixel positions across each row are shown

in the diagram on page 170.

So if all the eight available pixel positions (bits) in any one byte are used (that is, switched on or *enabled*) the corresponding decimal value of that byte is $128 + 64 + 32 + 16 + 8 + 4 + 2 + 1$, which is 255. If no pixels are used, the sum of the bit values is 0. Between the two extremes are unique values corresponding to any possible permutation of 'used' pixels. The value obtained can

■	WHAT IS A SPRITE?
■	DEFINING A SPRITE
■	FIRST STEPS
■	DATA GENERATION
■	USING THE SPRITE

■	SPRITES IN MEMORY
■	A NEAT TRICK FOR
	TRACING OFF YOUR DESIGN
■	AVOIDING SAVE AND LOAD
	PROBLEMS



over the sprite grid provided here or in the manual. Shown overleaf is an example of a sprite which actually forms part of the Space Station program on page 151.

This example shows the DATA values alongside.

Let's take a closer look at these. The first byte in the first row is not used—so the DATA value of these pixels is 0.

All but one of the pixels in the next byte are activated, with the unused pixel in the last place. The total value here is thus $128 + 64 + 32 + 16 + 8 + 4 + 2 = 254$. The third byte in the first row is again unused, contributing another 0 to the DATA. So the values for row 1 are therefore 0,8,0.

Before proceeding, have a look at the rest of the sprite's pixel arrangement. It is often possible to find another row which uses much the same configuration for one or more bytes, particularly in symmetrical designs. If this happens, you can save having to perform the DATA calculation more than once. In fact, there is no such match for row 1, but you will be able to do this with rows 14 and 16 or rows 17 and 19.

Now move on to the next row. In the first byte, pixels 2 and 1 are 'on', yielding a value of $2 + 1 = 3$. In the next byte, pixels 6,5,4 and 1 are 'on' to give $32 + 16 + 8 + 1 = 57$. And in the last byte only the first pixel (128) is used. Total for the row is thus 3,57,128.

In row 3, the first byte has three activated pixels and a total of $4 + 2 + 1 = 7$. The second byte is easy—all the pixels are on, for a maximum value of 255. The last byte has $128 + 64 = 192$. Check the values for the remaining rows in the same way.

With this done, you can group all the row values together to obtain a DATA statement. For the space station sprite this is:

```
20 DATA 0,254,0,3,57,128,7,255,192,0,
16,0,16,56,16,56,84,56,124,148,124,
131,255
30 DATA 130,144,58,18,184,16,58,144,16,
18,131,255,130,254,84,254,252,56,126,
0,56
40 DATA 0,0,40,0,0,56,0,1,199,0,6,16,
192,1,199,0,0,124,0
```

This is how this sprite appears in the Space Station program on page 151. But it isn't always necessary to write it in full this way. With suitable program adjustments so that these values aren't expected, it is usually possible to omit unused sprite rows at the start and end of a definition.

The remaining examples are less complicated designs, and calculating the values is much more straightforward. See if you can follow how the values are obtained, and whether you can find them in the DATA statements which are located in Lines 20 to 140 of the Space Station program.

DATA GENERATION PROGRAM

The least exciting part of working with a sprite is calculating the 63 DATA statement values needed to define it. Instead of doing this by hand, you can use a utility program to work out the values for you, and a wide choice of products of this type is available commercially.

But all you need to start with is a comparatively simple program which lets you 'draw' the sprite design on the screen, altering it as you wish. It will then automatically calculate the corresponding DATA values which you can use to define the sprite for use in other programs.

The program which follows does just this, and gives you the option, if you have a printer, of taking a hardcopy for future reference. The program is suitable only for single-colour high resolution sprites—but is nonetheless extremely useful. Although typing it in may seem like a long-winded way of doing one set of sprite calculations, when you have done it once you can SAVE the program for re-use every time you have a sprite to define:

```
10 POKE 53280,1:POKE 53281,1:DIM
A$(21),Z(3,21),A(24)
20 PRINT "▣ ▣"TAB(13)"SPRITE
EDITOR"
30 PRINT TAB(8):INPUT "PRINTER
OPTION.(Y/N)";I$:IF I$="Y"
THEN PR$="Y"
40 IF I$ <> "Y" AND I$ <> "N" THEN
GOTO 20
50 PRINT "▣ ▣"TAB(13)"PLEASE WAIT"
```

be used directly within a DATA statement as part of a sprite definition.

FIRST STEPS

Now let's look at some examples. As with ordinary UDGs, the best way to start is by designing the required shape on suitably ruled graph paper. Mark out an area containing 24 by 21 squares, and indicate the byte divisions. Alternatively, photocopy or trace

Row	128	64	32	16	8	4	2	1	128	64	32	16	8	4	2	1	128	64	32	16	8	4	2	1
0									0	254	0													
1									3	57	128													
2									7	255	192													
3									0	16	0													
4									16	56	16													
5									56	84	56													
6									124	148	124													
7									131	255	130													
8									144	58	18													
9									184	16	58													
10									144	16	18													
11									131	255	130													
12									254	84	254													
13									252	56	126													
14									0	56	0													
15									0	40	0													
16									0	56	0													
17									1	199	0													
18									6	16	192													
19									1	199	0													
20									0	124	0													

Sprite coding chart

An example of a sprite from the Space Station game—the station itself—shown superimposed on a grid giving row and bit numbers. See how the bit values can be added to form the DATA numbers. There are three numbers for each row because each one is made up of three bytes containing eight pixels each (see text). If you like, you can trace over this grid and use it when you are designing your own sprites and wish to work out the DATA

250 DATA“
260 DATA“
270 DATA“
280 DATA“
290 DATA“
300 DATA“
310 DATA“
320 DATA“
330 DATA“
340 DATA“
350 DATA“
360 DATA“
370 DATA“
380 DATA“
390 DATA“
400 DATA“
410 DATA“
420 DATA“
430 DATA“
440 DATA“
450 DATA“

```
60 FOR Z=1 TO 8:READA(Z):A(Z+8) =
  A(Z):A(16+Z) = A(Z):NEXT:DATA128,64,
  32,16,8,4,2,1
70 FOR Z=1 TO 21:READ A$(Z)
80 FOR ZZ=1 TO 8:IF MID$(A$(Z),
  ZZ,1) = "*" THEN Z(1,Z) = Z(1,Z) +
  A(ZZ)
90 NEXT ZZ:FOR ZZ=9 TO 16:IF
  MID$(A$(Z),ZZ,1) = "*" THEN
  Z(2,Z) = Z(2,Z) + A(ZZ)
100 NEXT ZZ:FOR ZZ=17 TO 24:IF
  MID$(A$(Z),ZZ,1) = "*" THEN
  Z(3,Z) = Z(3,Z) + A(ZZ)
110 NEXT ZZ,Z:PRINT " [ ] [ ] ":IF
  PR$ = "Y" THEN OPEN4,4:CMD4
120 PRINT TAB(12) " [ ] CHARACTER
  DATA [ ] [ ] "
130 FOR Z=1 TO 21:PRINT Z(1,Z);";";
  Z(2,Z);";";Z(3,Z);:IF Z < 21 THEN
  PRINT";";
140 NEXT Z:PRINT:PRINT:IF PR$ = "Y"
  THEN GOTO 170
```

```
150 PRINT TAB(7) " [ ] [ ] PRESS RETURN
  TO CONTINUE"
160 GET K$:IF K$ < > CHR$(13) THEN
  GOTO 160
170 PRINT " [ ] "TAB(11) " [ ] CHARACTER
  DESIGN [ ] [ ] "
180 PRINT TAB(7) " [ ] 7654321076543210
  76543210"
190 FOR Z=1 TO 21:PRINT TAB(7) " [ ] ";
200 FOR ZZ=1 TO 24:IF MID$(A$(Z),
  ZZ,1) = "*" THEN PRINT " [ ] [ ] ";
  GOTO 220
210 PRINT " . ";
220 NEXT ZZ:PRINT " [ ] ";Z:NEXT Z:IF PR$
  = "Y" THEN PRINT "#4, " [ ] ":
  CLOSE4
230 IF PR$ < > "Y" THEN GOTO 230
240 REM [ ] [ ] 765432107654321076543210
```


First design your flying bird on graph paper. It can then be entered as asterisks in the data generator program

To use the program, enter the instruction LIST 240-. This will display the 21 rows at the end of the program—empty DATA lines on which you can plot the shape of the sprite you want to design. Each character position within the DATA statement represents an available pixel position—the line is 24 characters wide. Use the cursor controls to move around the lines, then simply place an asterisk at every point of your design and remember to press **RETURN** to enter each line.

For guidance, the REM statement in Line 240 numbers each of the 24 horizontal positions which are available for each of the 21 vertical rows.

A typical design is shown bottom left. This shows the pattern of asterisks used to define the figure of a flying bird.

When you have completed your design, RUN the program. The screen then displays a prompt asking you whether direct printout is required. If you press N, output is to the



screen. First come the 63 DATA statement numbers—these you'll have to copy out by hand if you do not have a printer. For the example shown, the DATA looks like this:

```
CHARACTER DATA 128,0,3,192,0,30,240,
0,250,104,1,52,84,2,228,58,2,216,45,
7,144,20,133,32,22,78,64,22,44,64,11,
24,128,5,201,0,3,230,0,0,49,192,16,64,
32,31,129,16,21,194,240,27,54,8,10,9,0,
12,4,128,4,2,64
```

A large scale display of your sprite can then be obtained simply by pressing **RETURN**.

If you now choose to take a hardcopy using the printer, simultaneously press **RUN/STOP** and **RESTORE** and reRUN the program. Press Y when the printer option again shows. The printer output starts with a listing of all the character DATA, and this is followed by a large scale printout of the sprite on its entry grid.

Each sprite you define within this program may be **SAVED** as part of it. You can then **reLOAD** it at will and change it as necessary to create further sprites.

USING THE SPRITE

The design and DATA definition stage is perhaps the easiest part of using a sprite. Whichever way you work out your DATA values, they are of no use by themselves. It's only when you incorporate them in a new program that you can make use of the sprite which you have created.

First you need to store the sprite information somewhere suitable in the computer's memory. Then, you can call it up and manipulate it with your program.

Exactly how this is done is covered a little later. But for the moment, type in and RUN this program:

```
10 V = 53248:X = 150:Y = 157:PRINT "☐"
20 FOR I = 16000 TO 16062:READ A:POKE I,A:NEXT I
```



```
25 POKE 2040,250:POKE V + 21,1:GOTO 50
30 GET AS:A = 0:XX = 0:IF AS = "P" THEN A = 1:GOTO 50
35 IF AS = "L" THEN A = 2:GOTO 50
40 IF AS = "Z" THEN XX = - 2
45 IF AS = "X" THEN XX = + 2
50 FOR Z = 1 TO 10:X = X + XX:IF X > 250 THEN X = 30
55 IF X < 20 THEN X = 250
60 IF A = 1 AND Y > 70 THEN Y = Y - 2
65 IF A = 2 AND Y < 200 THEN Y = Y + 2
70 POKE V,X:POKE V + 1,Y
75 NEXT Z:GOTO 30
100 DATA 128,0,3,192,0,30,240,0,
250,104,1,52,84,2,228,58,2,216,45,
7,144,20
105 DATA 133,32,22,78,64,22,44,
64,11,24,128,5,201,0,3,230,0,0,49,
192,16,64
110 DATA 32,31,129,16,21,194,240,
27,54,8,10,9,0,12,4,128,4,2,64
```

This instructs the computer to set up a sprite and then move the image across the screen under keyboard control using the L, P, Z and X keys. In this case the sprite chosen is the flying bird figure defined earlier, but you can use the same program for any other single-colour sprite you invent—all you have to amend is the DATA statements at the end.

If you look at lines 100, 105 and 110, you'll see that these list the DATA values for the bird which were generated by the utility program.

Line 10 sets up a series of variables—basically a starting point for the sprite—and clears the screen. Line 20 **READS** the DATA which defines the sprite and holds it in the computer's memory. Line 30 then *enables* or turns the image on.

Lines 50 to 75 instruct the computer how to

Add the extra program line to the text program for a simple bird-zapping game using your newly defined sprite

```
15 POKE 53280,2:POKE 53281,2:POKE 650,
128:FOR Z = 16000 TO 16000 + 64*2:
POKE Z,0:NEXT
20 FOR I = 16000 TO 16077:READ A:
POKE I,A:NEXT I:TIS = "000000":
POKE V + 29,1
25 POKE 2040,250:POKE 2041,251:
POKE V + 21,3:XX = 100:YY = 100:
POKE V + 40,1
30 FOR Z = 1 TO 5:PRINT "☐☐TIME:
☐"VAL(TIS):IF RND(1) > .30 THEN
POKE V + 23,RND(1)*2
35 A = INT(RND(1)*3) + 1:X = X + 10:
IF X > 239 THEN X = 30
40 IF A = 1 AND Y > 70 THEN Y = Y - 10
45 IF A = 2 AND Y < 200 THEN Y = Y + 10
50 POKE V,X:POKE V + 1,Y:POKE V + 39,7:
GET Z$:IF Z$ = "Z" AND XX > 30 THEN
XX = XX - 5
55 IF Z$ = "X" AND XX < 250 THEN
XX = XX + 5
60 IF Z$ = "P" AND YY > 50 THEN
YY = YY - 5
65 IF Z$ = "L" AND YY < 220 THEN
YY = YY + 5
70 POKE V + 2,XX:POKE V + 3,YY:
IF PEEK(V + 30) = 3 THEN S = S + 1
75 PRINT "☐☐"TAB(25)"SCORE:
☐"S:NEXT Z:IF VAL(TIS) < 59 THEN 30
80 PRINT "☐☐TIMES UP ☐!":
END
115 DATA 8,0,0,8,0,0,62,0,0,8,0,0,8,0,0
```

Microtip

If you're using the sprite numbers/display program here, there's a quick way of transferring a design to the DATA statement lines of the program. Using tracing paper, mark off the key lines of your design—which should match the screen in size—and then tape this on the TV screen once you have **LISTed** Lines 240–450. Now simply use the cursor and plot asterisks 'underneath' the lines of your tracing until the arrangement of these closely matches the design on the tracing paper.

move the sprite under keyboard control. If you just want to display the sprite, try editing out Lines 40, 50, 55, 60, 70, and 75 from the program. In this case the bird will appear at a point fixed by the variables X and Y—which are set at 150 and 157, somewhere near the middle of the screen. If you enter alternative values for X and Y in Line 10, you can make the bird appear wherever you like.

The most complicated part of what's going on here is in Lines 10 to 25—the part that actually stores the sprite in the computer's memory. This is looked at in detail below. But don't worry if you don't understand it completely for the moment. How memory works is covered in depth in a later article.

SPRITES IN MEMORY

A sprite definition requires 63 bytes of memory for the 63 DATA values. But it is convenient to allocate 64 bytes for each, as this simplifies some of the calculations. These 64

bytes can be stored in any area of spare memory providing it is a multiple of 64.

Each of these definitions has to be located once it has been stored, and to do this *sprite pointers* are used. There are eight of these, and each can be **POKEd** with a value from 0 to 255. This value is multiplied by 64 to identify the sprite location. (This is the reason why the memory location has to be a multiple of 64.)

The maximum value—255—gives, therefore, a ceiling value of 255*64, which is 16K, covering the whole of one block of memory which the video chip can access. (There are four such blocks, or 16K *banks*—only one can be accessed at any one time.)

The pointers are a powerful tool which permit you to switch to and fro from any sprite definition which may be in memory. A whole sequence of sprites can be called in this way. Each one will replace its predecessor, so permitting imaginative animation effects.

In fact, this is a much more efficient way of

making use of sprites: switching the pointer rather than the sprites themselves leaves the sprites available for other uses.

The sprite pointers are always located at the unused end eight bytes of the 1K screen display memory, which is normally from 2040 to 2047. You can see an example of this in Line 30 of the flying bird program, where 2040 is used.

It is possible to relocate the screen memory and in so doing move the sprite pointers too. Fresh values for the pointers have to be **POKEd** into the new sprite pointer locations.

WHERE THE ACTION IS

Before you can control the sprite, you need to come to grips with the workings of the 6566 VIC-II chip used by the Commodore 64. In particular, you need to know how to access the control commands.

This isn't as fearsome as it sounds and is covered in greater detail in a later article. For the moment, take a glance at the following table, which lists the 47 memory addresses from 53,248 to 53,294 that are used in sprite programming. The commonly used V + value shows the relationship of all related addresses, based on the starting address where V = 53248.

At first sight the table seems rather daunting but it is nevertheless an essential reference point if you're going to get the most from your sprite programming. The V + value notation is rather easier to remember than the specific addresses, and is more efficient on memory usage—this would be especially beneficial in long programs. Additionally, their use considerably simplifies identification of specific sprite instructions—a useful aid when program changes and debugging are necessary.

You can see that each of the normal eight sprites is numbered, and the value of the number has important consequences.

You can also see from the table that there are locations for controlling everything from sprite positioning to collision detection and colour.

The actual use of V + value notation is described in much greater detail in the next part of this article which uncovers the inner workings of sprite control. But if you're curious, have a look at the use of V + values in the bird program.

There in Line 25 is V + 21, the enabling command which turns on the sprite. And in Line 70 the V values set up the X and Y positions of the sprite on the screen. Further examples which you can analyze by referring to the table appear in Line 2400 and others in the Space Station program.

VIC-II chip memory locations

This is a handy reference table of the memory locations you will need to access to control the shape, size, colour and position of your sprites.

The V + value numbers are the most useful and the easiest to remember.

Decimal	V + value	Description
53248	V	Sprite-0 X position
53249	V + 1	Sprite-0 Y position
53250	V + 2	Sprite-1 X position
53251	V + 3	Sprite-1 Y position
53252	V + 4	Sprite-2 X position
53253	V + 5	Sprite-2 Y position
53254	V + 6	Sprite-3 X position
53255	V + 7	Sprite-3 Y position
53256	V + 8	Sprite-4 X position
53257	V + 9	Sprite-4 Y position
53258	V + 10	Sprite-5 X position
53259	V + 11	Sprite-5 Y position
53260	V + 12	Sprite-6 X position
53261	V + 13	Sprite-6 Y position
53262	V + 14	Sprite-7 X position
53263	V + 15	Sprite-7 Y position
53264	V + 16	MSB of X coordinate
53265	V + 17	VIC control register
53266	V + 18	Raster register
53267		(light pen)
53268		(light pen)
53269	V + 21	Sprite display enable
53270	V + 22	VIC control register
53271	V + 23	Sprite 0-7 Y-expand
53272	V + 24	VIC memory control
53273	V + 25	Interrupt register
53274	V + 26	Interrupt enable
53275	V + 27	Background priority
53276	V + 28	Select multicolour
53277	V + 29	Sprite 0-7 X-expand
53278	V + 30	Sprite collision
53279	V + 31	Background collision
53280	V + 32	Screen border colour
53281	V + 33	Background colour 0
53282	V + 34	Background colour 1
53283	V + 35	Background colour 2
53284	V + 36	Background colour 3
53285	V + 37	Sprite multicolour 1
53286	V + 38	Sprite multicolour 2
53287	V + 39	Sprite-0 colour
53288	V + 40	Sprite-1 colour
53289	V + 41	Sprite-2 colour
53290	V + 42	Sprite-3 colour
53291	V + 43	Sprite-4 colour
53292	V + 44	Sprite-5 colour
53293	V + 45	Sprite-6 colour
53294	V + 46	Sprite-7 colour



TAPE SAVES

One of the unpublicised problems when working with sprites is attempting to **SAVE** programs on tape. This occurs only if you've made the quite forgivable error of actually **RUNning**, during the course of its development, a program containing a sprite. Quite normal, you might think, but a source of the mystifying **SAVE** errors rather too easily blamed on the C2N cassette unit!

The *only* fail-safe solution to the problem is to precede your normal **SAVE** command with the following **POKE**:

POKE 53269,0:SAVE "PROGRAM"


This disables the sprite display and eliminates the source of interference which affects the **SAVE** routine on tape. As an alternative, add the **POKE** to the end of the program itself.

The same **POKE** should be used to precede a **LOAD** instruction when a new program is **LOADed** into the Commodore following a sprite display. This is not necessary if you effect a system *coldstart*—usually done simply by switching the machine off and then on!

Disk users will be happy to know that sprite-related **SAVE** and **LOAD** problems do not extend to these units, and normal procedures may be used.

GET YOUR PROGRAMS IN SHAPE

- WHAT IS STRUCTURED PROGRAMMING?
- USING FLOW CHARTS
- HOW TO USE STRUCTURES IN BASIC PROGRAMS



Good design makes your programs easier to understand and easier to work on. It can also make the difference between them RUNNING efficiently or failing dismally

When you decide to write your first program for a computer there is usually an uncontrollable urge to sit down at the keyboard and start typing in some part of the program immediately. Maybe the first few lines you write work. You feel pleased with yourself and add a few more. You get these to work as well and so you go on adding lines here and there, testing as you go until the program has expanded to perhaps a few hundred lines.

Then, suddenly, disaster! You add a few

lines of code and it does not work any more. You cannot see any possible reason why it has stopped working. You change this line, that line, but still no joy.

However, do not give up. There are ways to stop this happening and all that it takes is a little organization. If you follow a few simple rules when you write your programs then you should have no trouble at all.

STRUCTURED PROGRAMMING

Standard BASIC was designed to be easy to use but it has very few *structures* compared to other languages and that makes it more difficult to write structured programs. Structures are the building blocks that you use to

construct the main shape of your program. In standard BASIC they are IF . . . THEN, FOR . . . NEXT, GOTO and GOSUB, and in BBC BASIC there are three extra ones—REPEAT . . . UNTIL, PROCedures and functions. You've seen how most of these are used on their own but the idea now is to put them together in an orderly and readable way.

It is very easy and quick to write a few lines of program that work first time (almost) and are relatively simple for someone else to understand. In fact it's not important to take special care structuring the program if it is short. The problem comes when you want to write a large program that does something useful. In this case, if you start by writing a program fragment and then add more and

more of the program to it you are likely to get hopelessly lost—unless you have the memory of an elephant, that is.

What you need to do is sit down and systematically design the program first. Most programs start with a vague idea of something you wish the computer to do—and the more complex this is, the vaguer your ideas are likely to be. The basic concept may be for a game or a word processing package or anything of a similar type which is so large that you cannot hold all the ideas about it in your head at once.

Start by finding somewhere well away from the computer—preferably in another room to avoid temptation—and write down what you want to do in a very general way. This is the beginnings of what is known as a *design specification* or *spec* for short. For example you might write:

Index system

The program should allow records in memory to be created, updated, deleted, sorted and listed. The records are to be accessed using a keyword. The entire set of records may be **SAVED** into and **LOAD**ed from a file.

Next you must break down this very general description into a few logical steps or *modules*. The operations involved in each module will probably still be very complicated so they must be broken down into smaller sections until you think that each of the lowest level of modules is short enough to handle. **Fig. 1** shows how this might be done for the Index System.

Each of the smallest sections should be no more than a page in length—about 60 lines—but half this length is even better. Each of these low level modules should be very easy to understand. Eventually each module will end up as a subroutine in your program.

The process of breaking problems down will become easier with experience. And, as always, the best way to learn how to do it is to try it. This method of breaking up the problem is known as top-down design. You started at the top (the general description of the program) and worked down to the bottom (the lowest level of modules). However, you still have not decided on the order of the modules—that is, the order in which they are executed by the program. So this comes next.

FLOW DIAGRAMS


One way to specify this is to use *flow diagrams* or *flow charts*. These are easy to use and being diagrammatical are easy to understand and follow at a glance. Like BASIC, they are not in themselves structured and so they should be used with caution. You can just as easily have a tangled and confused flow chart as you can have a jumbled up program. However, by

following a simple set of rules they may be used to sort out a program in a straightforward and structured way.


Fig. 2 shows how a single module can be specified. The program follows along the lines in the direction of the arrows and the boxes describe what happens at each stage. To execute a series of modules in sequence you just add on more boxes in the right order between start and stop (see **fig. 3**).

3. Combining several modules

2. A single module



1. Index system



A simple flow diagram like this is fine for a program in which no decisions are made. However, the power of the computer depends heavily on its ability to make choices within a program. This is the familiar IF ... THEN statement and it is about the most often used structure in BASIC.

Let's have a look now at all the different structures and how you can represent them with flow diagrams.

IF ... THEN ... ELSE

Although there are differences between the way different computers use the command, IF ... THEN ... ELSE is the basis of all decisions that the computer has to make. Its flow diagram is shown in **fig. 4** and in BASIC it is written as:

```
100 IF condition THEN statement1 ELSE
    statement2
```

It means, if the condition is true then statement 1 is executed, otherwise statement 2 is executed.

If you look at the flow diagram you'll notice that there is only one entry point and one exit. This greatly aids testing and debugging since you know exactly where this section of the program must start and finish.

In fact this is a very important rule of structured programming: for any section of code there should be only one entry and one exit.


Most versions of BASIC do not have the ELSE part—it isn't available on the Spectrum, ZX81, Commodore 64 or Vic. But that doesn't matter too much as it can be simulated using GOTO:

```
100 ...
110 IF condition THEN GOTO 140
120 statement2 :REM this is the ELSE part
130 GOTO 150
140 statement1 :REM this is the THEN part
150 ...
```

Of course the line numbers need not be as shown. Additionally, more than one statement may be included in the THEN and ELSE parts. For example, here's a section of program to sort two numbers into the correct order. It is the basis of an alphabetical sort routine which will be given in Part 2 of this article. The ELSE part in this case has four statements:

```
100 IF first <= second THEN GOTO 160
110 LET temporary = first
120 LET first = second
130 LET second = temporary
140 LET order$ = "wrong"
150 GOTO 170
160 LET order$ = "right"
170 ...
```

This example can also be written out using ELSE but then it must all be written on one line. Multiple



4. IF ... THEN ... ELSE

statements are allowed as long as they are separated by colons:

```
100 IF first > second THEN temporary =
    first: first = second: second = temporary:
    order$ = "wrong" ELSE order$ = "right"
```

As you can see, it's not easy to read or understand programs written using long statements like this. So they should be avoided if at all possible.

Finally, in many cases you may not need the ELSE part at all. The flow diagram looks like **fig. 5** and it is written out as:

```
100 IF condition THEN statement
    which, of course, is just the
    simple IF ... THEN statement.
```


NESTED STRUCTURES

The IF ... THEN ... ELSE lines may be nested. That is, one or both of the statements between which the IF ... THEN line chooses may themselves be an IF ... THEN line. For example, this section of program keeps tally of how many games two players have won, and PRINTs out the results after each game:

```
100 IF T1 < > T2 THEN GOTO 130
110 PRINT "It's a draw!"
120 GOTO 190
130 IF T1 < T2 THEN GOTO 170
140 PRINT "The first player wins"
150 LET P1 = P1 + 1
160 GOTO 190
170 PRINT "The second player wins"
180 LET P2 = P2 + 1
190 ...
```

All structures can be nested in any combination and, in theory, to any depth. However, the more nesting you have, the less readable the program becomes and a sensible limit is a depth of three or possibly four structures. If you find yourself needing more then you should split the program into smaller modules or subroutines.

Have a look at the last program again. It is actually rather difficult to follow what is going on even though it is perfectly well structured. A useful way to make nested



5. IF ... THEN

statements more readable is to indent the program lines. This is only possible on the Acorn and Spectrum computers but on these you can rewrite the last program as follows:

```
100 IF T1 <> T2 THEN GOTO 130
110 PRINT "It's a draw!"
120 GOTO 190
130 IF T1 < T2 THEN GOTO 170
140 PRINT "The first player wins"
150 LET P1 = P1 + 1
160 GOTO 190
170 PRINT "The second player wins"
180 LET P2 = P2 + 1
190 ...
```


On the Acorn computers you can automatically indent all the structured statements by typing LISTO 7 followed by LIST.

Other ways of making the structure of your program clearer are by inserting blank lines around different sections and by using REM statements. Blank lines can be entered on the Spectrum and Acorn computers by typing the line number followed by a space and then **ENTER** or **RETURN**. You can get a similar effect to this when you are writing programs on the Dragon, Tandy and Commodore by typing a colon instead of the space.

WHILE ... DO

The other essential structure is the WHILE ... DO statement. This allows looping in a program and is one of the most useful ways of creating a loop. It makes the computer DO something over and over again WHILE a certain condition is true. Don't worry that your computer doesn't have the words WHILE and DO. They are not available in most forms of BASIC. But you can make up a structure that does the same thing using IF ... THEN and GOTO.

The flow diagram is shown in **fig. 6** and in BASIC it is written out as:



6. WHILE ... DO loop

```
100 ...
110 IF NOT(condition) THEN GOTO 140
120 statement
130 GOTO 110
140 ...
```

Notice that Line 110 reads IF NOT(condition) ... That means it is checking for when the condition is NOT true—the opposite to normal. But there is no problem. If the condition is A=B then NOT(A=B) is A <> B—similarly NOT(A < B) is A > = B and so on. You can actually write NOT(A=B) if you like and the computer will understand what you mean.

Here is an example of a short program using a WHILE loop that you can use to time an egg:

```
5 CLS
10 PRINT AT 3,11;"EGG TIMER"
20 INPUT "How many minutes do you want?";t
30 PRINT AT 7,5;"Press any key to start"
40 PAUSE 0
50 CLS
60 PRINT FLASH 1;AT 10,9;
   "TIMING"
70 POKE 23672,0: POKE 23673,0
80 LET time = PEEK 23672 + 256*PEEK
   23673: IF time > t*50*60 THEN
   GOTO 110
90 PRINT AT 14,10;INT (time/50); "seconds"
100 GOTO 80
105 REM end of WHILE loop
```

```
110 PRINT FLASH 1;AT 14,10;"IT'S
   DONE"
120 BEEP .5,20
```



```
5 PRINT "TIMING":POKE 53281,4
10 PRINT TAB(16)"EGG TIMER"
20 INPUT "HOW MANY MINUTES DO YOU WANT";T
30 PRINT TAB(8)"PRESS SPACE
   BAR TO START"
40 GET AS:IF AS="" THEN GOTO 40
50 PRINT " "
60 PRINT TAB(15)"TIMING"
70 TIS="000000"
75 REM START OF WHILE LOOP
80 IF VAL(TIS) = > T*100 THEN GOTO 110
90 PRINT "TAB(15);
   RIGHT$(TIS,2)"SECONDS"
100 GOTO 80
105 REM END OF WHILE LOOP
110 PRINT TAB(16)"IT'S DONE"
120 POKE 54296,15:POKE 54278,128:
   POKE 54276,17:POKE 54273,50
130 FOR D=1 TO 200:NEXT:POKE
   54276,0:POKE 54278,0
```



```
5 PRINT "TIMING":POKE 36879,29
10 PRINT TAB(7)"EGG TIMER"
20 PRINT "HOW MANY MINUTES
   DO":INPUT "YOU WANT";T
30 PRINT "HIT SPACE BAR TO
   START"
40 GET AS:IF AS="" THEN 40
50 PRINT " "
60 PRINT TAB(8)"TIMING"
70 TIS="000000"
75 REM START OF WHILE LOOP
80 IF VAL(TIS) = > T*100
   THEN 110
90 PRINT "TAB(6);
   RIGHT$(TIS,2)"SECONDS"
100 GOTO 80
105 REM END OF WHILE LOOP
110 PRINT TAB(6)"IT'S
   DONE"
120 POKE36878,15:POKE36876,200
130 FOR D=1 TO 200:NEXT:POKE
   36876,0
```



```
5 CLS
10 PRINT TAB(15,2) "EGG TIMER"
20 INPUT TAB(5,4) "HOW MANY MINUTES
   DO YOU WANT";T
30 PRINT TAB(7,6) "PRESS SPACE BAR
   TO START"
40 KS = GET$
50 CLS:VDU23;8202;0;0;0;
60 PRINT TAB(15,10) "TIMING"
```



```

70 TIME = 0
75 REM start of WHILE loop
80 IF TIME > T*60*100 THEN
  GOTO 110
90 PRINT TAB(13,12);INT(TIME/100);
  " seconds"
100 GOTO 80
105 REM end of WHILE loop
110 PRINT TAB(14,15) "IT'S DONE"
120 VDU7

```



```

5 CLS
10 PRINT@43,"EGG TIMER"
20 PRINT@129,;:INPUT"HOW MANY
  MINUTES DO YOU WANT";T

```

7. Multiple choices

```

30 PRINT@196,"PRESS
  ANY KEY TO START"
40 AS = INKEY$:IF AS
  = "" THEN GOTO 40
50 CLS
60 PRINT@235,"TIMING"
70 TIMER = 0
75 REM START OF WHILE LOOP
80 IF TIMER > T*3000 THEN GOTO 110
90 PRINT@298,INT(TIMER/50);"SECONDS"
100 GOTO 80
105 REM END OF WHILE LOOP
110 PRINT@364,"IT'S DONE"
120 SOUND 180,3

```

MULTIPLE CHOICES

The IF ... THEN and WHILE ... DO structures are usually sufficient for most programs. However, there are a few extra structures to make programming even easier. For example, there are often more than two courses of action possible at a particular point in a program. This could be catered for using nested IF ... THENs but it is more convenient to use what is known as a CASE structure. This sets up each option in turn and directs the computer to a series of possible courses.


The flow diagram for a CASE structure is shown in **fig. 7** and in BASIC a typical

program would look like this:

```

100 REM CASE of command
110 IF CS$ = "C" THEN GOTO 170
120 IF CS$ = "U" THEN GOTO 190
130 IF CS$ = "D" THEN GOTO 210
140 IF CS$ = "L" THEN GOTO 230
150 PRINT "Command not recognised"

```



```

160 GOTO 240
170 PRINT "Create record": GOSUB 1000
180 GOTO 240
190 PRINT "Update record": GOSUB 2000
200 GOTO 240
210 PRINT "Delete record": GOSUB 3000
220 GOTO 240
230 PRINT "List records": GOSUB 4000
240 REM end of CASE

```

Another way of making multiple choices is by using the ON ... GOTO and ON ... GOSUB statements. However, when using ON ... GOTO, make sure that each of the options eventually directs you to the end of the section, as shown in the next example where a GOTO 1210 is needed after each option to direct the program to the end of the routine.

```

1000 REM POLYGON
  SUBROUTINE
1010 INPUT "HOW MANY
  SIDES DO YOU WANT";N
1020 ON N-2 GOTO
  1060,1100,1140,
  1160,1180,1200
1030 PRINT "I DON'T
  KNOW THE
  NAME OF A"
1040 PRINT "POLYGON

```

```


  WITH";N;
  "SIDES."
1050 GOTO 1210
1060 PRINT "THAT'S A TRIANGLE."
1070 PRINT "A TRIANGLE WITH EQUAL
  SIDES IS"
1080 PRINT "CALLED AN EQUILATERAL
  TRIANGLE."
1090 GOTO 1210
1100 PRINT "THAT'S A QUADRILATERAL."
1110 PRINT "A QUADRILATERAL WITH
  EQUAL SIDES"
1120 PRINT "AND ANGLES IS CALLED
  A SQUARE."
1130 GOTO 1210
1140 PRINT "THAT'S A PENTAGON."
1150 GOTO 1210
1160 PRINT "THAT'S A HEXAGON."
1170 GOTO 1210
1180 PRINT "THAT'S A HEPTAGON."
1190 GOTO 1210
1200 PRINT "THAT'S AN OCTAGON."
1210 PRINT
1220 RETURN

```

This is just a subroutine so you can't RUN it yet. The program to call the routine is given in the next section.

REPEAT ... UNTIL

The REPEAT ... UNTIL statement is another useful way of making a program loop (once



8. REPEAT ... UNTIL loop

again, don't worry if your computer hasn't got these commands—it can simulate them). In this case, unlike WHILE ... DO, the loop is always executed at least once. Have a look at the flow diagram in **fig. 8** and compare it to **fig. 6**. In BASIC it is written as:

```
100 ...
```



```

110 statement
120 IF NOT(condition) THEN GOTO 110
130 ...

```

Using the subroutine from the last example you can write a program using a REPEAT loop as follows:

```

10 PRINT "I WILL TELL YOU THE NAMES"
20 PRINT "OF SOME POLYGONS."
30 REM START OF LOOP
40 GOSUB 1000
50 INPUT "DO YOU WANT ANOTHER
   NAME ";AS
60 IF LEFT$(AS,1) = "Y" THEN GOTO 30
70 PRINT "GOODBYE!":END

```

Line 1000 is the polygon subroutine given in the last section.

Note that on some BBC computers you may need to change the semicolon to a comma in Line 50 of the program and Line 1010 of the subroutine.

BBC BASIC actually has a REPEAT ... UNTIL statement. What is more, the UNTIL part of the statement need not be on the same line as the REPEAT part, unlike the case of the IF ... THEN ... ELSE statement. So the last example can be written as:

```

10 PRINT "I will tell you the names"
20 PRINT "of some polygons."
30 REPEAT
40 GOSUB 1000
50 INPUT "Do you want another name ";
   AS
60 UNTIL LEFT$(AS,1) <> "Y"
70 PRINT "Goodbye!"

```

This is a lot easier to follow than the last version.

FOR ... NEXT LOOPS

You may not have realized it, but the familiar FOR ... NEXT loop is just a special case of the WHILE ... DO loop. It should be used when the number of times round the loop is known beforehand as this has to be specified at the start. The variable that keeps count of the number of times round the loop is known as the *control variable*.

A flow chart for a FOR ... NEXT loop looks something like **fig. 9**. Compare it to the WHILE loop in **fig. 6** and you'll see it has the same overall structure. In BASIC it is written:

```

100 FOR i = min TO max STEP val
110 statement
120 NEXT i

```

It is bad programming practice to jump out of a FOR loop using a GOTO statement since there is no way that the BASIC interpreter can know you have done this. You could quite legally jump back into the loop later on in the program, but understanding such a program would be hard. So don't do it!

PUTTING IT ALL TOGETHER

These structures are all you need, whatever type of program you are writing. But there is still a lot more to do before you can complete the program. You have to specify the variables you are going to use and make sure that the variables in the different modules don't clash with each other. Then you have to sort out what inputs and outputs are needed by the program.




Does the shape of the boxes used in flowcharts have any special significance?

Yes, flowcharts are drawn using standard symbols. There are five main shapes:

1. Rounded oblongs are called *terminal* boxes and they show where a program begins and ends.
2. Circles are *connecting* symbols used at the beginning and end of a module.
3. Rectangles are *instruction* boxes containing program statements.
4. Diamond shapes are *decision* boxes. There are always at least two routes out of these boxes depending on a decision made inside the box.
5. Finally, parallelograms (not shown in this article) are *input/output* boxes indicating any information input to the program, and any output to the screen or a printer.

Finally you have to test each of the modules and link them all together. How to do all this and still keep your program readable and structured will be covered in Part 2. Meanwhile, have a good, close look at some programs and see if you can identify the structures used—if you can find any that is! Then see if you can improve the program by making it more structured. Do this on your own programs or programs from magazines.



9. FOR ... NEXT loop

GETTING DOWN BELOW ZERO

- WHEN YOU NEED NEGATIVE NUMBERS
- NEGATIVE NUMBER CONVERSION PROGRAM
- THE SIGN CONVENTION



Binary and hex are easy to understand, and they translate directly onto the computer. But both have problems if you need to represent negative numbers.

In games programming you will find that you need to use negative numbers, for example, to move about the screen in certain directions. These have to be encoded in eight-bit bytes, since the computer has no other means to store anything in its memory. And this

raises a problem: as you have seen, a byte can represent any number between 0 and 255, or 00000000 and 11111111. But this uses up all the binary possibilities in eight bits—there is no room for a minus or plus sign and no way in which it could be represented.

In ordinary arithmetic, if you subtract 1 from 0 you get -1 . So try that calculation in eight-bit binary:

$$\begin{array}{r}
 00000000 \\
 \quad \quad -1 \\
 \hline
 11111111
 \end{array}$$

This does leave you trying to ‘borrow’ one from the next place to the left—but when the binary number is limited to eight bits there is nowhere left to borrow one from.

Similarly, if you subtract another 1 (to give -2 in ordinary arithmetic) you get 11111110. But if 11111111 in binary is 255 in decimal, then 11111110 must be 254!

It isn’t just in binary that you can get oddities like this. Imagine, for example, what would happen in decimal if you had no more than three places, or ‘columns’, in which to put your numbers. Look what happens when

you try to add 999 to 1000 when these limitations are imposed:

$$\begin{array}{r} 1000 \\ + 999 \\ \hline (1)099 \end{array}$$

The 1 in the 'thousands' place has had to go in brackets. In our three-place arithmetic, there is just no room for it. So the result of this addition, in the three-place system, is 99—which is exactly what you get if you subtract 1 from 1000!

Similarly, in a three-place system adding 998 to 1000 will act like subtracting 2. And subtracting 998 from 1000 will act like adding 2.

To sum up: the same row of numbers in an eight-bit byte can represent either a negative or a positive number—on the face of it, highly confusing and hard to work with.

What can you do about it? Well, for most home computer applications there is no need to worry. Memory addresses and operation codes, which are both rendered in binary, can be considered positive always. The only times you will come across negative numbers are in DATA or jumps, which are the machine code equivalent of GOTO statements.

FLIPPING THE BITS

In binary the process used to get from a positive number to its negative value is known as *2's complement*. It has no really solid theoretical basis—or at least none that is easy to understand—but it works.

To get from one binary number to its negative value, you flip the bits and add 1. 'Flip the bits' means change the bits that are 1 to 0 and those that are 0 to 1.

The following program shows you how this works. Note that when binary is limited to eight digits its hex equivalent exactly fills two digits. This means that the hex equivalent of the 2's complement also acts as the negative.

```

S
10 PRINT AT 0,7;"NEGATIVE NUMBERS"
20 PRINT AT 2,1;"DEC";TAB 14;
   "BIN";TAB 28;"HEX"
30 LET A$="□□□□□□□□"
   □□□□□□□□"
40 FOR N = 7 TO 13
50 PRINT AT N,6; INVERSE 1;A$
60 NEXT N
70 PRINT AT 8,8;"FLIP BITS";
   BRIGHT 1;AT 12,21;" + 1"
80 PRINT AT 10,3;" + ";AT 10,30;" + "
90 PRINT AT 15,8;"2'S COMPLEMENT"
95 LET C = 0
100 LET DD = -C: DIM A(8)

```

```

110 PRINT AT 4,0;"□□□□";AT
   4,4 - LEN STR$ C;C
115 POKE 23608,C: LET E = PEEK 23608:
   LET Z = E: GOSUB 300: PRINT AT
   4,29;A$
120 PRINT AT 17,0;"□□□□";AT
   17,4 - LEN STR$ DD;DD
130 LET D = 128: LET CC = E
140 FOR N = 1 TO 8: LET A(N) = 0
150 IF CC - D > = 0 THEN LET A(N) = 1: LET
   CC = CC - D
160 PRINT BRIGHT 1;AT 4,6 + 2*N;A(N);
   AT 10,6 + 2*N;1 - A(N)
170 LET D = D/2: NEXT N
180 POKE 23608,DD: LET DD = PEEK
   23608: LET D = 128
185 LET Z = DD: GOSUB 300: PRINT
   AT 17,29;A$
190 FOR N = 1 TO 8: LET B = 0: IF
   DD - D > = 0 THEN LET B = 1: LET
   DD = DD - D
200 PRINT BRIGHT 1;AT 17,6 + 2*N;B:
   LET D = D/2: NEXT N
210 PRINT AT 18,0;"-----□□"
   "-----□□-----"
220 PRINT AT 19,3;"0□□□□□"
   "0□0□0□0□0□0□0□0□0□0"
   "□□□□□□00"
230 IF INKEY$ = "" THEN GOTO 230
240 LET A$ = INKEY$: IF A$ = "□" THEN
   LET C = C + 1: IF C = 128 THEN LET
   C = -128: BEEP 1,1
250 IF A$ = "B" OR A$ = "b" THEN LET
   C = C - 1: IF C = -129 THEN LET
   C = 127: BEEP 1,1
260 IF A$ < > "□" AND A$ < > "B" AND
   A$ < > "b" THEN INPUT "?";C
270 GOTO 100
300 LET ZA = INT (Z/16): LET
   ZB = Z - (16*ZA)
310 LET ZA = ZA + 48: IF ZA > 57 THEN
   LET ZA = ZA + 7
320 LET ZB = ZB + 48: IF ZB > 57 THEN
   LET ZB = ZB + 7
330 LET A$ = CHR$ ZA: LET A$ = A$ + CHR$
   ZB: RETURN

```

```

S
10 PRINT AT 0,7;"NEGATIVE NUMBERS"
20 PRINT AT 21, 1;"DEC"; TAB 14;
   "BIN";TAB 28;"HEX"
30 LET A$ = "■ ■ ■ ■ ■ ■ ■ ■"
   "■ ■ ■ ■ ■ ■ ■ ■"
40 FOR N = 7 TO 13
50 PRINT AT N,6; A$
60 NEXT N
70 PRINT AT 8,8;"FLIP BITS□";AT
   12,21;" + 1"
80 PRINT AT 10,3;" + "; AT 10,30;" + "
90 PRINT AT 15,8;"2S COMPLEMENT"


```

```

95 LET C = 0
100 LET DD = -C
105 DIM A(8)
110 PRINT AT 4,0;"□□□□□";
   AT 4,4 - LEN STR$ C;C
115 POKE 16507, C
116 LET E = PEEK 16507
117 LET Z = E
118 GOSUB 300
119 PRINT AT 4,29;A$
120 PRINT AT 17,0;"□□□□□";
   AT 17, 4 - LEN STR$ DD;DD
130 LET D = 128
135 LET CC = E
140 FOR N = 1 TO 8
145 LET A(N) = 0
150 IF CC - D > = 0 THEN LET A(N) = 1
155 IF CC - D > = 0 THEN LET CC =
   CC - D
160 PRINT AT 4,6 + 2*N;A(N);AT 10,
   6 + 2*N; 1 - A(N)
170 LET D = D/2
175 NEXT N
180 POKE 16507,DD
181 LET DD = PEEK 16507
182 LET D = 128
185 LET Z = DD
186 GOSUB 300
187 PRINT AT 17,29;A$
190 FOR N = 1 TO 8
191 LET B = 0

```





```

10 CLS
20 PRINT@8,"NEGATIVE NUMBERS";
30 PRINT@40,STRING$(16,
  CHR$(131));
40 PRINT@65,"DEC"TAB(15)"BIN"
  TAB(28)"HEX"
50 PRINT@226,"+"TAB(29)"+";
60 PRINT@361,"2'S COMPLEMENT"
70 PRINT@483,"00000000
  0000000000000000
  0000";
80 FOR J=1474 TO 1502:POKE
  J,131:NEXT
90 FOR J=1 TO 7
100 FOR K=1 TO 24
110 POKE 1123+K+32*J,175
120 NEXT K,J
130 PRINT@165,"FLIP BITS";
140 PRINT@313,"+1";
150 AT=AT AND 255
160 T=AT:IF T=128 THEN SOUND 30,2
170 LN=3:GOSUB280:GOSUB310
180 T=T+256*(T>127):GOSUB340
190 T=255-T:LN=7:GOSUB280
200 T=T+1:T=T AND 255:LN=13:
  GOSUB280:GOSUB310
210 T=T+256*(T>128):GOSUB340
220 IN$=INKEY$:IF IN$ <> "B" AND
  IN$ <> " " AND IN$ <> CHR$(13)

```

In the sign convention, 128 acts as its own negative, and the computer starts counting upwards again

```

      THEN 220
230 IF IN$="B" THEN AT=AT-1:
  GOTO150
240 IF IN$=" " THEN AT=AT+1:
  GOTO150
250 PRINT@384,;:INPUT AT
260 PRINT@384,"0000";
270 GOTO 150
280 FOR X=0 TO 7
290 IF-(T AND 2↑X) THEN PRINT
  @LN*32+23-X*2+(X>3),"1"; ELSE
  PRINT@LN*32+23-X*2+(X>3),"0";
300 NEXT:RETURN
310 IF T<16 THEN AS$="0" ELSE
  AS$=""
320 PRINT@LN*32+29,AS$+HEX$(T);
330 RETURN
340 PRINT@32*LN,MID$("0000"+
  STR$(T),LEN(STR$(T)));
350 RETURN

```

SIGN CONVENTION

For most purposes, as mentioned above, a binary or hex number can act perfectly well both as a positive and a negative number. But sometimes you need to know whether a

number is positive or negative.

When you want a program jump in machine code you have to specify by how many bytes you want the computer to jump with a positive number, to jump forwards; a negative number for backwards.

What the computer does is to look at the first bit of the binary number and decide for itself whether the number is negative or positive. If the first bit is 1, the computer takes it as negative. If it is 0, the computer takes it as positive.

This process is known as the *sign convention*. It means that, instead of taking the eight-bit binary numbers to mean 0 to 255, the computer treats them as -128 to +127.

In the 2's complement conversion program, you will notice that the computer beeps when you get to 128. This is because 128, which is 10000000 in binary, or 80 in hex, acts as its own negative. (You can check for yourself: -10000000 + 10000000 = (1)00000000 or 0 in eight-bit binary. Likewise 80+80=(1)00 or 0 in two digit hex. And 128-128=0 in decimal.)

So which is it to be? As 10000000 has a 1 in its first bit, the computer treats it as a negative number: -128. On the other hand zero—or 00000000—has a 0 in its first bit. So the computer treats it as positive.

REFINING YOUR SCREEN GRAPHICS

Home computers offer plenty of scope to the budding artist. Here are some ways to expand your use of the BASIC graphics commands and create new screen pictures.

Once you have mastered the basics of drawing images on the screen, you can start to extend your artistic efforts by using some of the specialized graphics commands available on your computer. Commands like MOVE, PLOT, DRAW, PAINT and CIRCLE allow free rein to your imagination and allow you to create anything from a display chart to brighten up a business program, to the background for an exciting adventure game.

The article on page 84 showed how to use simple drawing commands to create line pictures on the screen and, in some cases, how to add colour. But you can extend these basic techniques with new ways of plotting dots, drawing lines, triangles, squares and circles. Used by themselves or in combination with colours, they become a powerful means of forming static or even moving images.

The way in which different computers use these commands varies from one to another, but each is capable of some really interesting effects. The exceptions are the Commodore 64 and Vic 20, whose standard BASIC does not include these commands. However, you can gain them by the addition of a suitable ROM cartridge (see page 87), and this article contains a section explaining how you can use this facility to extend your Commodore's drawing potential.

CIRCLES AND ARCS

Circles and arcs are among the most useful Spectrum 'tools' for drawing on-screen static graphics.

This golf course program shows how to use them to draw trees, fences, water, rough and bunkers, as well as describing the principles behind them.

You can check your progress as you go by RUNNING each group of lines. Do not NEW your machine each time; if you leave the lines intact, you will end up with the scene in **fig. 1**.

Before you start on the circles, however, it is best to get the clubhouse out of the way. So start by entering these lines:

```
90 BORDER 4: PAPER 4:CLS
200 LET w=10: LET s=50
210 FOR c=162 TO 174
```

```
220 PLOT INK 2;w,c
230 DRAW INK 2;s,0
240 LET w=w+2: LET s=s-4
250 NEXT c
260 FOR b=148 TO 162
270 PLOT INK 2;10,b
280 DRAW INK 2;50,0
290 NEXT b
295 DRAW INK 2;10,-3: DRAW 0,-11
300 PRINT INK 0;AT 2,2; "■"; AT 2,4;
"■"; AT 2,6; "■"
```

Lines 200 to 250 DRAW the roof, using techniques similar to those in the earlier chapter on plotting and drawing (pages 84 to 86). It starts off at 10, 162 on the screen with a line which is 50 pixels wide. Then its width decreases by four pixels for every one pixel that it rises.

A similar loop in Lines 260 to 290 DRAWS the walls, with two extra lines (in Line 295) to form a porch. Then Line 300 fills in the windows by the simplest possible method—PRINTing a black square from the standard ROM graphics characters through the walls at three places.

DRAWING AN ARC

On the Spectrum, as explained earlier (page 86) the easiest way to draw a complete circle is to use the CIRCLE command.

But if you want only part of a circle, the easiest way is to use an addition to a conventional DRAW statement. You can get a huge variety of effects by varying this statement, so it is worthwhile conducting a few experiments before you go any further. For example, try:

```
10 PLOT 130, 30
20 DRAW 0, 10, 1
30 GOTO 20
```

(Don't worry about the error report.)

As you'll recall, the first two numbers in Line 20 tell the computer to DRAW a line from the PLOTted point to a spot 10 pixels above it. What the last number is doing is to make that line curve, rather than going in a straight line. How much of a curve you get is dictated by the size of the number. This tells the Spectrum to DRAW part of a circle. A whole circle is represented by 2 times π so the



number 1 makes it DRAW about one-sixth of a circle (or, more accurately, 1 divided by 2 times π). If you try changing Line 20 to:

```
20 DRAW 0, 10, 2
```

... you'll find you get a more pronounced series of curves. Similarly, 0, 10, 3 gives a sawtooth pattern, 0, 10, 4 gives part of a chain-link fence (or half the trunk of a palm tree, depending on how you view it!), while 0, 10, 6 produces a spiral.


If, however, you try

```
20 DRAW 0, 10, 2*PI
```

... you may get a surprise. What the Spectrum wants to do is DRAW a circle two of

- USING MOVING LINES ON THE DRAGON AND ACORN
- GOLF COURSE SCENE FOR SPECTRUM AND ACORN
- LETTERS ON THE DRAGON

- MORE WAYS WITH COLOUR
- PICTURES FROM CIRCLES AND ARCS
- SIMON'S BASIC FOR COMMODORE GRAPHICS



whose points are in a straight line. But since such a circle would be rather bigger than your living room—indeed, rather bigger than the solar system—it **DRAWs** only that part of it that it can.

For drawing scenery, try this:

```
10 PLOT 0, 100
20 DRAW RND*5 + 5, 0, 2
30 GOTO 20
```

This **DRAWs** the wave pattern of a choppy sea. For calmer waters, try $RND*10 + 10$ or even $RND*15 + 10$.

One point to remember about such arcs is that a *negative* number at the end of your **DRAW** line will produce the arc's mirror image. Now clear Lines 10 to 30 and you can continue building up the golf course scene.

FENCE AND LAKE

The golf course program has two examples of arcs used for graphic effect—the first for the little fence in front of the clubhouse; the second for the lake.

First type in these lines:

```
310 FOR f=0 TO 84 STEP 3
320 PLOT f,142
330 DRAW 3,0,-3
340 NEXT f
345 DRAW 35,-42: DRAW -12,-6
```

Here the starting off point is 0, 142 on the screen. What Line 330 does is to **DRAW** a series of tight arcs—semi-circles only three pixels wide—to form the fence. The number you get is governed by the **FOR . . . NEXT** loop.

Next, enter and **RUN** these lines:

```
100 LET x=130: LET y=125: LET z=50
110 PLOT INK 5;x,0
120 DRAW INK 5;y,z,-1.25
130 LET x=x+1: LET y=y-1: LET z=z-1
140 IF x>254 THEN GOTO 170
150 IF z<1 THEN LET z=0
160 GOTO 110
```

The picture here is more complicated. First the computer **PLOTs** a point 130 pixels from the left and 0 pixels from the bottom of the screen. Then it **DRAWs** a line to a point 125 pixels to the right, and up 50 pixels from the

bottom of the screen, 'bending' the line by -1.25 radians as it goes.

From that stage the variables take over. Variable x starts each line one pixel farther to the right, y makes the line one pixel shorter than the previous one (else it would run off the screen), while z makes the line end one pixel lower than the previous line.

Eventually, of course, z would become a minus number, making the computer try (but fail) to PRINT off the bottom of the screen. Hence the need for Line 150, which makes all the short lines towards the end of the program finish on the bottom line of the screen.

TREES AND BUSHES

The golf course program also uses complete circles—as a substitute for a PAINT or similar statement, available on some computers but not on the Spectrum—to produce trees and scrub. These few lines give some bushes randomly behind the 'green':

```
400 FOR r=172 TO 168 STEP -1
410 LET x=RND*45+195
415 PLOT x,r-2: DRAW 0,-2
420 CIRCLE x,r,RND*2+1
440 CIRCLE x+10,r,RND*2+1
450 NEXT r
```

While these few give similar scrub on the right-hand side:

```
460 FOR r=135 TO 172 STEP 6
470 LET y=252
480 CIRCLE y,r,RND+2
490 NEXT r
```

The trees in the left-hand bottom corner are too big to be DRAWn randomly. So a different technique, READ ... DATA (see pages 104 to 109) is used instead:

```
900 FOR w=1 TO 3
910 READ a,b
920 PLOT a,b
930 DRAW 0,-24
940 LET f=RND*5+5
950 CIRCLE a-10,b+f,f: CIRCLE a,b+f,
f: CIRCLE a+10,b+f,f
960 CIRCLE a-5,b+f*2,f: CIRCLE a+5,
b+f*2,f
970 CIRCLE a,b+f*3,f
980 NEXT w
3000 DATA 20,70,52,85,84,100
```

The trick here is to DRAW the trunks first, working downwards from the original PLOTting points, so that you do not get ugly trunk marks showing through the 'foliage'. The trunks are PLOTted at 20, 70 and so on by the DATA in Line 3000. Line 940 randomizes the sizes of the foliage, while the $b+f$ in Line

950 makes sure that the bottom rows of circles begin a suitable distance above the trunks.

FINISHING TOUCHES

The driving tee (not exactly pukka; more the rubber mat you might find on a municipal course!) is DRAWn by these lines:

```
1000 LET t=30
1010 FOR y=0 TO 10
1020 PLOT t,y
1030 DRAW -30,30
1040 LET t=t+2
1050 NEXT y
```

And these few lines give the flags on the greens:

```
170 PLOT 220,140
180 DRAW 0,15: DRAW 8,-3: DRAW -8,-2
190 PLOT 22,120
195 DRAW 0,18: DRAW 9,-3: DRAW -9,-2
```

Finally, you'll need some bunkers. The neatest—but a very slow—way of DRAWing these, in the absence of a PAINT statement, is to start with a tiny ellipse and make it grow, pixel by pixel, until it reaches a suitable size.

How to DRAW ellipses is explained in detail in a later article on the computer's mathematical functions. In the meantime, enter these lines, then go off and have a coffee while they are RUNNING:

```
1495 LET r=1
1500 FOR x=0 TO 2*PI STEP PI/180
1510 PLOT INK 6;168+r*SIN
x,147+r*COS x/2.5
1520 PLOT INK 6;235+r*SIN
x,106+r*COS x/2.75
1530 PLOT INK 6;225+r*SIN
x,97+r*COS x/2.5
1540 NEXT x
1550 LET r=r+2
1560 IF r>20 THEN GOTO 6000
1570 GOTO 1500
```

Alternatively, you might like to make up simpler, but cruder, bunkers using UDGs!



With the addition of the Simons' BASIC cartridge a large number of additional programming commands are made available on the Commodore 64. Some of these are provided to simplify graphics programming and the use of several of these has already been explained (see pages 87–88).

ARC and ANGL are two extra drawing commands not yet covered. The first is used to draw parts of the circumference of a circle and the command takes the form:

```
99 ARC 150,50,60,270,1,30,30,1
```

This would normally be preceded by the HIRES command, so enter this line also:

```
90 HIRES 0,1
```

The first pair of figures after the ARC command define the screen coordinates of the centre of the circle whose arc you require. As usual, these values refer to the pixel positions in the standard configuration of horizontal (X value) first, vertical (Y value) next.

Try adjusting one or both these figures and reRUN the program several times. You'll see that the higher each of these values becomes, the closer to the right or bottom of the screen the curve is plotted until you eventually reach the border.

The actual length of the arc is regulated by the next two pairs of values. The first pair (60 and 270) are the start and end angles. If you imagine you're dealing with a clockface, the angle count begins and ends at 12 o'clock with the values 0 and 360. Thus our example curve begins at 2 o'clock and ends at 9 o'clock. Once again, try changing the values of just this pair of figures to gauge the effect.

The next figure (1) is the plotting increment and by adjusting this to a value in the range 1 to 360 you choose the interval in degrees the curve produced using the value 10 will look somewhat coarser than that of the original program line. As an indication of the possibilities, make these amendments to the original program:

```
90 HIRES 0,1: FOR N=10 TO 360 STEP 10
99 ARC 160,80,0,360,N,84,60,1
100 NEXT : PAUSE 10
```

This shows the effect of adjusting the plotting increment in steps of 10 degrees for a curve which starts and ends at the same point—in other words, a circle. (The use of variable, N, in this example suggests a powerful method of ringing the changes with minimal extra programming.)

The pair of values which follows N in this example regulates the physical appearance of the 'circle' whose arc is displayed. The first value is the length, in pixels, of the X or horizontal radius; the second is that for the Y or vertical radius. If you want a circular arc the X value must be 1.4 times the Y value in HIRES mode, and 1.6 times in MULTI mode. Otherwise an elliptical curve instead of a proper circle is the result.

Try adjusting these two values but remember that you have a maximum resolution of 320 pixels horizontally, 200 vertically in HIRES mode, and half the horizontal resolution in MULTI mode. The sum of the very first value after the ARC command—the X coordinate of the curve centre—and the value

of the X radius of the curve must not exceed the horizontal limit. Similarly, the sum of the a problem: as you have seen, a byte can exceed 200. In both cases, exceeding the limit causes the curve to run along the screen edge until the curve reaches an allowable value. At this point it continues on its way again.

The final figure in the command line is the *plot type* value. Putting a 0 here would clear a dot, obviously of no use. Putting a 1 plots a dot, and a 2 inverts a dot (turns 'off' a dot that is 'on', and 'on' a dot that is 'off'). Substitute the value 2 in the program above to see what happens.

DRAWING RADII

The next command, ANGL, is used for drawing the radius of a circle and can be used in a variety of ways to depict things like the spokes of a wheel or the blades of a fan. It takes the form (NEW the computer):

```
10 HIRES 0,1: FOR N=0 TO 360 STEP 4
20 ANGL 160,80,N,84,60,1
30 NEXT: PAUSE 10
```

Again, to make the program do a lot of work with little programming, a variable is included as part of a FOR ... NEXT loop. The first pair of values after the ANGL command are the X and Y coordinates of the centre of the circle whose radius is to be drawn—the starting point of the radius, of course. The value N is the *angle* to the perpendicular at which the radius is to be drawn. Thus the value 45 here would draw a line from the start point to another at 3 o'clock to it. As you can see, values of 0 to 360 can be used. Try changing the STEP values from 1 to 10, or setting a fixed value for N in Line 20.

The remaining three values serve exactly the same function they do with the ARC command. But you can usefully look at the 'unplot' value 0. Add these lines to your existing program:

```
30 NEXT
35 CIRCLE 160,80,84,60,1
40 FOR N=360 TO 0 STEP -4
45 ANGL 160,80,N,84,60,0
50 NEXT: PAUSE 10
```

This repeats the pattern, adds a circle and Line 45 then proceeds to erase the work of Line 20.

BLOCKS OF COLOUR

The command BLOCK enables you to create coloured rectangles. You could do this by using REC and PAINT but BLOCK is often more convenient, particularly where several rectangles are required. The command takes the form that follows (NEW your computer first):

```
10 HIRES 0,1: MULTI 2,5,6
20 BLOCK 10,10,30,30,1
50 PAUSE 10
```

When RUN, this displays a single colour rectangle. The first pair of values define the





PLOT and DRAW commands don't just limit you to simple straight line drawing of the type covered in detail on page 88. With the right controls, curves, zig-zags and a whole range of textures are at your disposal. And with new uses for colour as well, you have the basis for a whole range of visual effects.

The article on page 84 showed how you can get to grips with simple drawing using PLOT and DRAW—and how to add colour to your pictures with just a few more instructions. Once you've mastered the basic commands, you can add a new dimension to your graphics by introducing lines whose shape can be controlled as they are being drawn. You can add waves to the surface of your seas, put jagged rocks on the outline of a cliff face, or build battlements on a castle. All of these can be done with a single line.

The secret of adding texture with PLOT and DRAW commands is not to be limited by the fixed coordinates that these specify to draw a single line, but to use the computer's speed to draw 100 or more short lines, each at a different angle, across the screen. And instead of one zig-zag line, you can draw 50 or even 2000 at different heights across the screen to produce a wide range of different effects. For example, a program could move the cursor a random distance up and down for every unit that it moves to the right. This would draw an irregular jagged line—just what you need for drawing rocks, or perhaps a cracked pane of glass.

DRAWING RANDOM LINES

One example of this type of control uses FOR ... NEXT loops to alter the co-ordinates of a

TROUBLE SHOOTER

Unlike most computers, the Acorn machines allow you to use the whole of the TV screen for your text or drawing. However, on some TV sets you might find that the display is too far up or too far down the screen so that you lose part of the picture. This commonly happens near the corners of the screen.

Luckily, the BBC computer has a way to correct this using the *TV command. *TV 1 moves the display up one line and *TV 255 moves it down one line. Other numbers between 1 and 255 move the display even more. Typing a ,1 after it turns off the interlace (fuzziness).

But note that these commands only come into effect after you press **BREAK** or change MODE.

line randomly between limits. To see how this works, enter the lines below and RUN the program. Do not be put off by the % signs; they are there just to make the plotting faster. The program would work without them.

```
10 MODE 2
20 GCOL 0,1
30 LET Y%=800
40 FOR X%=0 TO 1279
50 LET Y%=Y%-30
60 IF Y% < 0 THEN END
70 MOVE X%, Y%
80 FOR A%=1 TO 150
90 PLOT 1,RND(20),RND(30)-15
100 NEXT A%,X%
```

What this does is to give a series of lines across the screen. To achieve this, a colour mode (Line 10) and a colour in which to draw (Line 20) are selected.

The crux of the program is Line 90, which draws a line from a point on the left edge of the screen to a random point to the right, then from a point near this new point to another random point, and so on across the screen. The starting point is set up at Lines 30 to 50. These lines set X%=0 and Y%=770. Line 90 lets the cursor move to the first random point and PLOTs. The next starting point is selected at Lines 40 and 50, and the next finishing point is selected at Line 90.

The effect will vary according to the number of times each line is drawn and the limits you set for A% in Line 80. The variations on this routine are numerous. Try changing each value in turn to see the result. You might want a perspective view, to give the effect of a mountain range—as you will see when the next block of program is RUN. It is almost the same as the one above, but with different values:

```
10 MODE 2
20 GCOL 0,1
30 LET Y%=-180
40 FOR X%=-10 TO 1279
50 LET Y%=Y%+20
60 IF Y% > 150 THEN END
70 MOVE X%,Y%
80 FOR A%=1 TO 150
90 PLOT 1,RND(20),RND(40)-15
100 NEXT A%,X%
```

DRAWING AN AREA OF ROUGH

Alternatively, you might want just a small patch of the screen textured by drawing lines



1. Use arcs and circles for the Spectrum golf course



2. Simons' BASIC makes drawing easy on the Commodore

across the television screen. The next block of program could represent a growth of tall grass or weeds, just the thing for an area of rough ground—in a game of golf, for example. Type NEW, then type in and RUN these next few lines:

```
10 MODE 2
20 GCOL 0,130:GCOL 0,1
30 CLG
40 FOR T=1 TO 50
50 MOVE 600+RND(100),600+RND(100)
60 FOR A=1 TO 10
70 PLOT 1,RND(20),RND(30)-15
80 NEXT A,T
```

Line 30 clears the screen to green—the background colour selected at Line 20. This time, though, the starting points are randomly selected (Line 50) each time the program looks at Line 40. The finishing points, too, are randomly selected and PLOTted (Line 70) each time the program looks at Line 60. This gives an irregular edge to the textured area, instead of a sharp rectangular edge. Again, the effect can be varied extensively. Try the program in MODE 1, then change Line 10 back to MODE 2.

MAKING A BUNKER

The next element in a golf course might be a

sand obstacle or bunker. The technique of PLOTting random co-ordinates is again a useful solution to this problem. Add these lines to the program above:

```
90 GCOL 0,3
100 FOR T=0 TO 150
120 MOVE 200+RND(100),50+RND(100)
130 FOR A=0 TO 10
140 PLOT 65,20,RND(50)-10
150 NEXT A,T
```

This time the PLOT command displays not lines, but single points. As for the rough, a random starting position is set up (Line 120). Each time the program steps through the FOR ... NEXT loop starting at Line 100, a single point is PLOTted 20 units away in the X direction and -9 to 40 units in the Y direction.

Here is another picturesque example of the dot shading techniques:

```
200 MODE 2
210 GCOL 0,130
220 CLG
230 GCOL 0,3
240 FOR Y%=1024 TO 1000 STEP -8
250 FOR X%=0 TO 1279 STEP RND(5)+10
260 IF RND(1200)<Y% THEN PLOT 69,X%,Y%
270 NEXT X%,Y%
```

Line 210 selects a green background, then Line 260 builds up a picture of shaded dots. The striking perspective of the whiter background is given by the IF ... THEN condition at Line 260, which PLOTs points less frequently as the program advances.

This type of control is ideal when you wish to make a picture or set the scene for a game. And you can bring several elements together or reposition them simply for different views. For example, here are some lines to bring together the rough, bunker and two woods:

```
200 GCOL 0,0
210 FOR Y%=1023 TO 850 STEP -10
220 MOVE 0,Y%
230 DRAW 150,Y%
240 NEXT Y%
245 REM. ....WOODS 2
```




```

250 FOR Y%=840 TO 700 STEP -10
260 MOVE 0,Y%
270 LET X%=Y%-690
280 DRAW X%,-20,Y%
290 NEXT Y%
295 REM. ....WOODS 3
400 LET X1%=500
410 LET X2%=1100
420 FOR Y%=1023 TO 825 STEP -10
430 LET X1%=X1%+10
440 MOVE X1%,Y%
450 LET X2%=X2%-20
460 DRAW X2%,Y%
470 NEXT Y%

```

Notice that as you enter these lines they will overwrite the previous lines for the picture.

COMPLETING THE GOLF COURSE

Enter the next block of lines to change an area of the background colour to form a lake:

```

500 GCOL 0,4
510 MOVE 1279,650
520 MOVE 1279,600
530 PLOT 85,1100,600
540 PLOT 85,1000,400
550 MOVE 1279,600
560 PLOT 85,1279,100

```

This block of program PLOTs and fills three triangles—the first is set up at Lines 510, 520 and 530, the second at Lines 520, 530 and 540 and the third at Lines 540, 550 and 560. Enter the next block and RUN the program to see a colourful course complete with a flag to mark one of the holes and a rectangle to mark a building:

```

590 REM. ....BUILDING
595 GCOL 0,1
600 MOVE 0,500
610 MOVE 50,500
620 PLOT 85,0,300
630 PLOT 85,50,300
640 REM. ....FLAG
700 MOVE 300,800
710 DRAW 300,900
720 GCOL 0,6
730 MOVE 380,880
740 PLOT 85,300,860

```

Both objects are PLOTted as triangles—in a similar way to before.

Building up the golf course in stages like



this lets you test and refine each element before adding it to the whole picture. Now you know how it's done, see if you can design a different picture, like the snow scene that is illustrated on page 189.



THE DRAW COMMAND

The LINE and CIRCLE commands are very useful for drawing simple or regular shapes with little effort by the programmer. But as you try to draw more sophisticated shapes you'll find that your programs become very long and unwieldy. Imagine how long the program would be to draw something like the ship in fig. 1, if you had to use a LINE statement for each separate line on the ship.

To avoid this problem, you need to use the computer's DRAW command. Using DRAW, you can direct the course of a line while it is being drawn. You can tell the computer to take the line a set distance to the right, then some other distance up, then to the left, and so on. You give it these instructions as a string, so the program can be much more compact.

As an example, type in and RUN this program, which creates the ship in only ten lines:

```

10 PMODE 4,1
20 PCLS5
30 SCREEN1,1
40 DRAW"BM23,96C0"
50 DRAW"R28E2U3L6UR6E2R5F2D3R3U2R7D
  2R4U9E2R4F2D5R3U2R4U6E3R5D8"
60 DRAW"R3U24L3UR8DL4D24RF2R5D4R4U4
  R6U9E3R5D10R4U2R4U14RD10R3U2"
70 DRAW"R4D11R7U3E2R4F2D3RD8R3U5E2R
  8D2R6U2R7UE2R6F2D4R4U4E2R5F2"
80 DRAW"R6DL6D3R24G12L195H4U5"
90 PAINT(127,100),0,0
100 GOTO 100

```

The program works like this:

Lines 10 to 30 set up the initial conditions. Line 10 selects PMODE 4 so that the highest resolution can be used. PCLS 5 in Line 20 clears the screen and changes it to Buff.

The series of DRAW commands in Lines 40 to 80 form the outline of the ship. In each case, DRAW operates on the string of instructions, which are contained within the inverted commas which follow.

Line 40 is the simplest, with a short string that contains instructions telling the machine where to start DRAWing, and in what colour. The first instruction is the letters BM. This stands for Blank Move, and positions the 'drawing pen' at the place you want your graphic to start. The starting position is given by the co-ordinates which follow it, in this case 23,96—notice that these aren't enclosed

by brackets as in some of the other graphics commands. The final instruction in the string is the DRAW colour, C0 or Black.

Line 50 starts to trace the outline. The string which controls the DRAW command may look very confusing, but it's really very simple. It consists of a series of directions and distances. The letters control the direction of the line and the numbers give its length in pixels. If there isn't a number after a direction, then it will move one pixel only.

There are eight directions which you can use: U means Up, D means Down, L means Left, R means Right, E means draw at 45 degrees (up to the right), F means draw at 135 degrees (down to the right), G means draw at 225 degrees (down to the left) and H means draw at 315 degrees (up to the left).

Reading from the start of Line 50, the strings tells the computer to draw right 28 pixels, at 45 degrees 2 pixels, up 3 pixels, and so on. Try translating the string into pencil movements on graph paper to see how the ship's outline builds up.

Lines 60 to 80 contain similar strings which complete the ship's outline. You could use just one long string instead of all these lines, but a large number of instructions tends to get unwieldy and hard to correct. Finally, the outline is turned into a silhouette by Line 90 which PAINTs the ship black.

DRAWING LETTERS

One of the limitations with graphics programs is that the Dragon and Tandy can't display text on the graphics screen. This means that you can't PRINT out a score, for example, on a game that uses the high resolution graphics. The game on page 98, for example, is all displayed on the text screen for that reason.

There is a solution to this problem, though. You can design your own characters using the DRAW command. Type in and RUN this program and you'll see how HELLO can be DRAWn:

```

10 PMODE 3,1
20 PCLS
30 SCREEN1,0
40 HES="D4BR3U2NL3U2BR5L3D2NR2D2
  R3BR5L3U4BR5D4R3BR2U4R3D4L3"
50 DRAW"BM110,50;C3S8"+HES
60 GOTO 60

```

The message is DRAWn in PMODE 3, a four-colour mode.

Line 40 shows you another characteristic of the DRAW command. As it operates on a string of instructions you can define string variables and call them up in a DRAW command later in the program.

The instructions in HES tell the computer

how to DRAW the letters in HELLO. Try translating the instructions into pen movements as you did before, remembering that B means blank, and no line will appear from DRAWing with an instruction preceded by a B.

If you have several words you wish to display on the graphics screen, such as BAD LUCK, WELL DONE, and so on, you can set up further strings, BL\$ and WD\$ for example. Try working out the instructions for one of these on graph paper. Once you have defined the strings, you can call these up whenever they are needed in the program.

HELLO is DRAWn by Line 50. The start position is at 110,50, the colour is number 3 (Blue) so you'll see a blue word, and in size 8 (double scale). S may be followed by a number from 1 to 62—1 is quarter size, 4 is normal size (the *default*, which you get in the absence of instructions) and 8 is double size.

As Line 50 demonstrates, you can join up strings of DRAW instructions just like any other strings. If you want to call up other strings—such as BL\$ or WD\$—you can set them up in just the same way as Line 50. Move to the start point using BM, then set up the colour with C and the size with S.

LONGER MESSAGES

Defining strings for each message you want to display can become tedious if you have a large number of them.

The solution is to define all the characters you might want to use, by typing in a program like the one that follows:

```

10 PMODE 3,1
20 DIM LE$(26)
30 PCLS
40 FOR K = 0 TO 26:READ LE$(K):NEXT
50 FOR K = 0 TO 9:READ NU$(K):NEXT
60 DATA BR2,ND4R3D2NL3ND2BE2,ND4R3
  DGNL2FDNL3BU4BR2,NR3D4R3BU4BR2,
  ND4R2FD2GL2BE4BR,NR3D2NR2D2R3
  BU4BR2
70 DATA NR3D2NR2D2BE4BR,NR3D4R3U2
  LBE2BR,D4BR3U2NL3U2BR2,ND4BR2,
  BD4REU3L2R3BR2,D2ND2NF2E2BR2
80 DATA D4R3BU4BR2,ND4FREND4BR2,
  ND4F3DU4BR2,NR3D4R3U4BR2,ND4R3
  D2NL3BE2,NR3D4R3NHU4BR2
90 DATA ND4R3D2L2F2BU4BR2,BD4R3U2
  L3U2R3BR2,RND4RBR2,D4R2U4BR2,
  D3FEU3BR2,D4EFU4BR2
100 DATA DF2DBL2UE2UBR2,DFND2
  EUBR2,R3G3DR3BU4BR2
110 DATA NR2D4R2U4BR2,BDEND4BR2,
  R2D2L2D2R2BU4BR2,NR2BD2NR2BD2
  R2U4BR2,D2R2D2U4BR2,NR2D2R2D2L2
  BE4,D4R2U2L2BE2BR2,R2ND4BR2,
  NR2D4R2U2NL2U2BR2,NR2D2R2D2U4BR2

```

```

120 SCREEN1,0
130 A$ = "TESTING 0123456789"
140 DRAW "BM60,50;C3S8"
150 GOSUB 9000
160 GOTO 160
9000 FOR K=1 TO LEN(A$)
9010 B$ = MID$(A$,K,1)
9020 IF B$ >="0" AND B$ <="9"
  THEN DRAW NU$(VAL(B$)):GOTO 9050
9030 IF B$ = " " THEN N = 0 ELSE
  N = ASC(B$) - 64
9040 DRAW LE$(N)
9050 NEXT
9060 RETURN

```

A character set, consisting of capital letters from A to Z, digits from 0 to 9 and, most important, a space, is contained in the DATA lines—Lines 60 to 110. This DATA is READ into the letter and number arrays—LE\$ and NU\$—by Lines 40 and 50.

To use the character set, you must define the message, along with the start position and any other information, such as size and colour. Line 130 contains a test message, A\$. In this case it is TESTING 0123456789, but you could substitute anything you choose. Line 140 sets the start position, the colour and the size. With the message defined and the start point set, you can tell the machine to DRAW the message. The printing subroutine—starting at Line 9000—examines each character in A\$, one at a time, finds the appropriate instructions in the arrays (see Games Programming, pages 144–147) and then DRAWs the character on the screen.

By changing the string in Line 130, and, if necessary, the start position in Line 140 you can DRAW any message you wish. Since the part of the program that DRAWs the letters is a subroutine, you can have as many messages as you like during the program—you must call them all A\$ though.

This is extremely useful if, for example, you have a series of messages you wish to display at various points during a game. You could incorporate the letter drawing program into the game as it stands—start the game at Line 120 or somewhere after the DATA, except for any DIM, CLEAR and PCLEAR lines which should be right at the start of the whole program. When you want to write the message, just set up strings containing the display you want, at the point it is needed in the program. Then add a DRAW statement like in Line 140, and call the printing subroutine by including GOSUB 9000.

AND FINALLY

Once you've worked out the instructions for drawing a shape you can draw it the right way

up, on either side, or even upside down. All you need to do is to include a further instruction in the string, A followed by a value from 0 to 3.

A0 means DRAW at 0 degrees—in other words upright. A1 will put the house on its side, leaning to the right—at 90 degrees. A2 will place the house upside down—at 180 degrees. And A3 will put the house on its side, leaning to the left—at 270 degrees.

Being able to control the angle at which the image is DRAWn is useful because you can DRAW a graphic in four different ways from just one set of instructions—you do not have to tell the computer how to DRAW each separate version of the graphic.

To see how it works in practice, type in and RUN this program:

```

10 PMODE 3,1
20 PCLS
30 SCREEN 1,0
40 S$ = "NR16E8F4U4R2D6F2D12L6U6L4
  D6L6U12"
50 FOR K=1 TO 20
60 D = RND (200) + 27:E = RND (140) + 27:
  C = RND (3) + 1:A = RND (4) - 1
70 DRAW "BM" + STR$(D) + "," +
  STR$(E) + "C" + STR$(C) + "A" +
  STR$(A) + "XSS;"
80 NEXT K
90 GOTO 90

```

You will find 20 houses, each identical except for their colour and orientation.

The shape of the houses is defined in the string in Line 40. Four random numbers are chosen by Line 60—D and E are the start co-ordinates, C is a colour and A is the angle at which the house will be DRAWn.

Line 70 DRAWs the house, adding together all the parts of the string and the random numbers. The STR\$ function converts the numeric variables into strings—in effect it puts inverted commas round the value of the numeric variables. For example, if D = 2 then STR\$(D) = "2".

What Line 70 does, then, is to begin with a blank move to a random start point, add a random colour instruction, and finally the DRAWing instructions themselves at a random angle. The X before S\$ means 'execute a substring', and allows you to add one string to another during DRAWing rather like +HE\$ did earlier.

The time to use X is when string space is limited in the machine's memory. X allows you to put a string together from various parts without actually creating a new string—adding strings together requires more string space, which wastes memory.

CUMULATIVE INDEX



An interim index will be published each week. There will be a complete index in the last issue of *INPUT*.

<p>A</p> <p>AND 35-36</p> <p>Animation 26-32</p> <p>Applications</p> <ul style="list-style-type: none"> family finance 136-143 hobbies' files 46-53, 75-79 letter writer 124-128 <p>Assembly language 66-67</p> <p>Assignment statement 66-67, 92</p> <p>ATTR, Spectrum 68-69</p> <p>B</p> <p>BASIC 65</p> <p>BASIC programming</p> <ul style="list-style-type: none"> arrays 152-155 decision making 33-37 how to PLOT, DRAW, LINE, PAINT 84-91 inputting information 129-135 programmer's road signs 60-64 READ and DATA 104-109 random numbers 2-7 refining your graphics 184-192 screen displays 117-123 structured programming 173-178 the FOR...NEXT loop 16-21 variables 92-96 <p>Binary 38, 41, 44, 45, 113-116</p> <ul style="list-style-type: none"> negative numbers 179-183 <p>Brackets, use of 35</p> <p>Breaking out of a program 4</p> <p>Bridge, drawing a</p> <ul style="list-style-type: none"> <i>Spectrum</i> 108 <p>Byte, definition of 114</p> <p>C</p> <p>Cassette recorders, choice of 24</p> <p>Castle, drawing a</p> <ul style="list-style-type: none"> <i>Dragon, Tandy</i> 108 <p>Christmas program</p> <ul style="list-style-type: none"> <i>Acorn</i> 64 <p>CHRS, Dragon, Tandy 26-27</p> <p>CIRCLE 86-91</p> <p>CLEAR</p> <ul style="list-style-type: none"> <i>Dragon, Tandy</i> 14, 27 <i>Spectrum</i> 10 <p>Clock, internal 69-73</p> <p>COLOUR 87-90</p> <p>Control variables 94</p> <p>Craps program 63</p> <p>Cursor, definition of</p> <ul style="list-style-type: none"> control codes, <i>Commodores</i> 123 <p>D</p> <p>DATA</p> <ul style="list-style-type: none"> for arrays 154-155 for graphics 107-109 machine code 67 statements 8-14, 40-45 <p>Decimal</p> <ul style="list-style-type: none"> conversions from binary 38, 42 converting fractions into binary 114 <p>Decision making 33-37</p> <p>Default colours</p> <ul style="list-style-type: none"> <i>Acorn</i> 89 <i>Dragon, Tandy</i> 90 <p>Delays in programs 17</p> <p>DIMensioning an array 152-153</p> <p>DRAW 85-91</p> <p>Drawing letters, Dragon, Tandy 191-192</p> <p>E</p> <p>Egg-timer program 176-177</p> <p>ENDPROC, Acorn 64</p> <p>Error, causes of 36</p>	<p>F</p> <p>Family finance program 136-143</p> <p>File, saving and loading a 77</p> <p>Filing system program 46-53, 75-79</p> <p>Flow charts 173-178</p> <p>Flying bird sprite, Commodore 64 168-172</p> <p>FOR...NEXT loop 16-21</p> <p>G</p> <p>Games</p> <ul style="list-style-type: none"> aliens and missiles 144-151 animation 26-32 arrays for games 155 bombing run program 161-167 controlling movement 54-59 firing missiles 55-58 fruit machine 36 guessing 3-5 maze game 68-74 minefield 97-103 moving characters 54-59 routines 8-15 scoring and timing 69-73 space station game 144-151 visual explosions 161-167 <p>GET, Commodore 64 55, 132-134</p> <p>GET\$, Acorn 55, 57, 58, 103, 132-134</p> <p>GET#, Commodore 64, Vic 20 135</p> <p>Golf-course, drawing a</p> <ul style="list-style-type: none"> <i>Acorn, Spectrum</i> 184-191 <p>GOSUB 62-64</p> <p>GOTO 18-21, 60-62</p> <p>Graph-drawing program</p> <ul style="list-style-type: none"> <i>Acorn</i> 64 <p>Graphics</p> <ul style="list-style-type: none"> characters 38-45 creating and moving UDGs 8-15 drawing on the screen 132-133 drawing pictures 107-109 explosions for games 161-167 fire-breathing dragon 80-83 frog UDG 10-15 instant embroidery 21 'low-resolution' painting by numbers 19 refining your graphics 184-192 sunset pattern 20 tank UDG 10-15 using PLOT, DRAW, CIRCLE, LINE, PAINT 85-90 also see animation; movement; ROM graphics; teletext; UDG. <p>H</p> <p>Helicopter, building a</p> <ul style="list-style-type: none"> <i>Commodore 64</i> 31 <p>Hexadecimal 38, 42, 45, 156-160</p> <p>HIRES, Commodore 64 87</p> <p>Hobbies file 46-53, 75-79</p> <p>House, drawing a</p> <ul style="list-style-type: none"> <i>Acorn</i> 107-108 <i>Commodore 64</i> 108-109 <p>I</p> <p>IF...THEN 3, 33-37</p> <p>INK, Spectrum 86</p> <p>INKEY, Acorn 28-29, 103, 134-135</p> <p>INKEY\$, Spectrum 54-55, 132-135</p> <p>INPUT 3-5, 117-122, 129-135</p> <p>INPUT prompts 130-131</p> <p>INT, Commodore 64, Spectrum 2-3</p> <p>K</p> <p>Keypress, detection of 54-55</p> <p>Keywords, spelling of 19</p>	<p>L</p> <p>Languages, computer 65</p> <ul style="list-style-type: none"> see Assembly language; BASIC; Machine code <p>Letter writing program 124-128</p> <p>LINE, Dragon, Tandy 88-91</p> <p>Line numbers, in programs 7</p> <p>Logical operators 35-37</p> <p>Lower case letters, Dragon, Tandy 142</p> <p>M</p> <p>Machine code</p> <ul style="list-style-type: none"> advantages of 66 binary numbers 113-116 drawing dragon with 80-83 games graphics 38-45 hexadecimal 156-160 low level languages 65-67 negative numbers 179-183 nonary numbers 111-112 number bases 110-116 speeding up games routines 8-15 <p>Maze programs 68-75</p> <p>MIDS, Acorn 71</p> <ul style="list-style-type: none"> <i>Commodore 64</i> 101-102 <p>Minefield game 97-99</p> <p>MODE, Acorn 28</p> <p>MOVE, Acorn 71, 88-90</p> <p>Movement 26-32, 59</p> <p>N</p> <p>Negative binary numbers, conversion program 180-183</p> <p>Nested loop, definition and use of 19</p> <p>NEW 10-15, 23</p> <p>Nonary numbers 111</p> <p>Null strings 96</p> <p>Number bases 110-116</p> <p>O</p> <p>ON...GOSUB 64</p> <p>ON...GOTO 62</p> <p>Opcodes 67</p> <p>Operators 35</p> <p>OR 35-36</p> <p>P</p> <p>Parameters 64</p> <p>Password program 133</p> <p>PAUSE</p> <ul style="list-style-type: none"> <i>Commodore 64</i> 88 <i>Spectrum</i> 101, 108 <p>PEEK 59, 101</p> <p>Peripherals, cassettes 22-25</p> <p>Pixel 84</p> <p>PLOT 88-89</p> <p>PMODE, Dragon, Tandy 90</p> <p>POINT, Acorn 71</p> <p>POKE</p> <ul style="list-style-type: none"> <i>Commodore 64</i> 15, 99, 108-109 <i>Dragon, Tandy</i> 13, 40, 101 <i>Spectrum</i> 101 <p>Positioning text 117-123</p> <p>PRINT 26-32, 117-123</p> <p>PRINT AT</p> <ul style="list-style-type: none"> <i>Dragon, Tandy</i> 26-27 <i>Spectrum, ZX81</i> 8-9, 31-32 <p>PRINT TAB, Acorn 11, 28</p> <ul style="list-style-type: none"> <i>Commodore 64, Vic 20</i> 30 	<p>PROCedures, Acorn 64</p> <p>Program</p> <ul style="list-style-type: none"> BASIC 8 BREAKing into 4, 7, 11 line numbers 7 punctuation of 4 slowing down 17 <p>PSET, Dragon, Tandy 13, 90-91</p> <p>Punctuation, in PRINT statements 119-123</p> <p>R</p> <p>RAM 25, 44, 46</p> <p>Random numbers 2-7</p> <p>READ 40-44, 104-109</p> <p>REC, Commodore 64 87</p> <p>Records 75-77</p> <p>REPEAT...UNTIL, Acorn 36</p> <p>Resolution, high and low 84</p> <p>RESTORE 106-107</p> <p>RETURN 62</p> <p>RIGHT\$, Commodore 64 101, 102</p> <p>RND function 2-7</p> <p>ROM graphics 26-32, 107-109</p> <p>Running man, building a, Acorn 28-29</p> <p>RUN/STOP, Commodore 64, Vic 20 7</p> <p>S</p> <p>Satellite, building a</p> <ul style="list-style-type: none"> <i>Dragon</i> 26-27 <p>SAVE 22-25</p> <p>Scoring 97, 100-101</p> <p>SCREEN, Dragon, Tandy 40, 90</p> <p>Screen drawing program 132-133</p> <p>Screen formatting 117-123</p> <p>Shell, firing a 10-15</p> <p>Ship, drawing a</p> <ul style="list-style-type: none"> <i>Dragon, Tandy</i> 191 <p>Simons' BASIC, Commodore 64 87-88</p> <p>Snow scene, Commodore 64 186-188</p> <p>Spaces, using</p> <ul style="list-style-type: none"> <i>Commodore 64, Vic 20</i> 122 <p>Sprite, Commodore 64 14, 15, 168-172</p> <p>STEP 17, 21</p> <p>STOP, Spectrum, ZX81 4, 64</p> <p>String variables 4-5, 95-96</p> <p>STRING\$ 98</p> <p>Subroutines 62-63</p> <p>Symbols, arithmetic 6</p> <p>T</p> <p>TAB 117-122</p> <p>Tables, multiplication 5-7</p> <p>Tank, controlling and creating a 10-15</p> <p>Teletext graphics, BBC 28</p> <p>Terminating numbers 34</p> <p>Timing 97, 101-103</p> <p>Twos complement 179-183</p> <p>U</p> <p>UDG, definition of 8-15, 40-44</p> <ul style="list-style-type: none"> grids for 8-11 DATA for 45 creating your own 38-45 <p>V</p> <p>VAL, Commodore 64 101</p> <p>Variables 3-5, 92-96, 104-108</p> <p>VDU command, Acorn 28-29, 70, 99</p> <p>Verifying saved programs 24-25</p> <p>VIC chip memory locations</p> <ul style="list-style-type: none"> <i>Commodore 64</i> 172
--	---	---	---

The publishers accept no responsibility for unsolicited material sent for publication in INPUT. All tapes and written material should be accompanied by a stamped, self-addressed envelope.

COMING IN ISSUE 7 ...

- ❑ Discover how the **MEMORY** of your computer is built up—and what you are doing when you load machine code programs into it
- ❑ In **Games Programming**, there's a run-down on how to add **LEVELS OF DIFFICULTY** to make your games suitable for beginners or experts—plus a complete new maze game program
- ❑ Learn about **STRING SLICING** techniques which give you the ability to make use of stored information selectively
- ❑ To bring together your skills at **STRUCTURING A PROGRAM**, there's an analysis of the techniques applied to a complete program for numeric or alphabetical sorting
- ❑ **JOYSTICKS** are among the cheaper and more versatile of the peripherals you can add—and they're not just limited to games



ASK YOUR NEWSAGENT FOR INPUT